A Machine-checked Proof of Birkhoff's Variety Theorem in Martin-Löf Type Theory

William DeMeo
Thmpr Research

Jacques Carette
McMaster University

Abstract

The Agda Universal Algebra Library (agda-algebras) is a library of types and programs (theorems and proofs) we developed to formalize the foundations of universal algebra in dependent type theory using the Agda programming language and proof assistant. In this paper we draw on and explain many components of the agda-algebras library, which we extract into a single Agda module in order to present a self-contained formal and constructive proof of Birkhoff's HSP theorem in Martin-Löf dependent type theory. In the course of our presentation, we highlight some of the more challenging aspects of formalizing the basic definitions and theorems of universal algebra in type theory. Nonetheless, we hope this paper and the agda-algebras library serve as further evidence in support of the claim that dependent type theory and the Agda language, despite the technical demands they place on the user, are accessible to working mathematicians (such as ourselves) who possess sufficient patience and resolve to formally verify their results with a proof assistant. Indeed, the agda-algebras library now includes a substantial collection of definitions, theorems, and proofs from universal algebra, illustrating the expressive power of inductive and dependent types for representing and reasoning about general algebraic and relational structures.

2012 ACM Subject Classification Theory of computation → Logic and verification; Computing methodologies → Representation of mathematical objects; Theory of computation → Type theory

Keywords and phrases Agda, constructive mathematics, dependent types, equational logic, formalization of mathematics, Martin-Löf type theory, model theory, universal algebra

Funding *William DeMeo*: supported by the CoCoSym project under ERC Consolidator Grant No. 771005.

1 Introduction

The agda-algebras library is a repository of types and programs (theorems and proofs) formalizing the foundations of universal algebra in Martin-Löf dependent type theory (MLTT) using the Agda programming language. The library now includes an fairly extensive collection of formal definitions, theorems, and proofs that codify, in the formal language of type theory, the analogous definitions, theorems, and proofs of classical, set-theory-based universal algebra and equational logic. As such, the agda-algebras library provides many examples that exhibit the expressiveness of inductive and dependent types for representing and reasoning about general algebraic and relational structures in a formal language. The main advantage of formalizing mathematics in type theory using a proof assistant (like Agda) is that the software checks the correctness of our proofs by a process known as "type-checking."

The first major milestone of the agda-algebras project is a formal proof of *Birkhoff's variety theorem* (also known as the *HSP theorem*) [4] in dependent type theory. Our first formal proof of the theorem, completed in January of 2021, contained some flaws and there were concerns that the proof was not truly constructive.[1] We are confident that the version

[1] See the Birkhoff module from the 15 Jan 2021 commit (71f1738) of the ualib/ualib.gitlab.io repository [6].

we present here—based on version 2.0.0 of the agda-algebras library—is fully constructive and correct.[2] To the best of our knowledge, ours is the first formulation of the HSP theorem in MLTT, and the first formal, machine-verified proof of Birkhoff's celebrated 1935 result.

In this paper, we present a self-contained formal proof of the HSP theorem by extracting into a single Agda module a subset of the agda-algebras library, including only the pieces we need for the proof. The main body of the paper is generated by a literate Agda file, available online,[3] that others can type-check, using Agda version 2.6.2 and Agda Standard Library version 1.7, to verify its correctness. We include here every line of code of our formal proof of Birkhoff's theorem in a single, self-contained (apart from a few dozen imports from the Agda Standard Library) Agda module.

In the course of this presentation we highlight some of the challenging aspects of formalizing the basic definitions and theorems of universal algebra in type theory. One positive contribution of this project is that it lends support to the claim that dependent type theory and the Agda language, despite the technical demands they place on the user, are accessible to working mathematicians (such as ourselves) who possess sufficient patience and resolve to codify their work in type theory in order to formally verify their results with a proof assistant.

Our presentation gives a sobering glimpse of the technical hurdles that must be overcome to conduct research in mathematics using dependent type theory and the Agda language. Nonetheless we hope our work does not discourage anyone from investing in these technologies and we remain committed to the use and promotion of type theory and proof assistants in general and in our own research. Indeed, we are excited to share the gratifying outcomes and achievements that resulted from attaining some degree of mastery of type theory, interactive theorem proving, and the Agda language.

2 Preliminaries

2.1 Logical foundations

An Agda program typically begins by setting some language options and by importing types from existing Agda libraries. The language options are specified using the OPTIONS *pragma* which affects the way Agda behaves by controlling the deduction rules that are available and the logical axioms that are assumed when the program is type-checked to verify its correctness. Every Agda program in the agda-algebras library, including the module Demos.HSP described in this paper,[4] begins with the line {-# OPTIONS –*without-K* –*exact-split* –*safe* #-}. Here are brief descriptions of these options, accompanied by links to related documentation.

- *without-K* disables Streicher's K axiom. See the section on axiom K in the Agda Language Reference Manual [16].
- *exact-split* makes Agda accept only those definitions that behave like so-called *judgmental* equalities. See the Pattern matching and equality section of the Agda Tools documentation [19].
- *safe* ensures that nothing is postulated outright—every non-MLTT axiom has to be an explicit assumption (e.g., an argument to a function or module). See the cmdoption-safe section of [17].

[2] Specifically, see the 30 Nov 2021 commit (ab859ca) of the agda-algebras library [7].
[3] See `https://github.com/ualib/agda-algebras/blob/master/src/Demos/HSP.lagda`
[4] available at `https://github.com/ualib/agda-algebras/blob/master/src/Demos/HSP.lagda`

The OPTIONS pragma is usually followed by the start of a module and a list of import directives. For example, the present module (Demos.HSP) begins as follows.

```
-- Import universe levels and Signature type (described below) from the agda-algebras library.
open import Algebras.Basic using ( 𝓞 ; 𝓥 ; Signature )

module Demos.HSP {S : Signature 𝓞 𝓥} where

-- Import 16 definitions from the Agda Standard Library.
open import Data.Unit.Polymorphic                  using ( ⊤ ; tt                    )
open import Function                               using ( id ; flip ; _∘_           )
open import Level                                  using ( Level                     )
open import Relation.Binary                        using ( Rel ; Setoid ; IsEquivalence )
open import Relation.Binary.Definitions            using ( Reflexive ; Symmetric     )
                                                   using ( Transitive ; Sym ; Trans  )
open import Relation.Binary.PropositionalEquality  using ( _≡_                       )
open import Relation.Unary                         using ( Pred ; _⊆_ ; _∈_          )

-- Import 23 definitions from the Agda Standard Library and rename 12 of them.
open import Agda.Primitive renaming ( Set    to Type    ) using ( _⊔_ ; lsuc                )
open import Data.Product   renaming ( proj₁ to fst     )
                           renaming ( proj₂ to snd     ) using ( _×_ ; _,_ ; Σ ; Σ-syntax   )
open import Function       renaming ( Func  to _⟶_     ) using ( Injection ; Surjection     )
open       _⟶_             renaming ( f     to _⟨$⟩_   ) using ( cong                       )
open       Setoid          renaming ( refl  to reflˢ   )
                           renaming ( sym   to symˢ    )
                           renaming ( trans to transˢ  )
                           renaming ( _≈_   to _≈ˢ_    ) using ( Carrier ; isEquivalence   )
open       IsEquivalence   renaming ( refl  to reflᵉ   )
                           renaming ( sym   to symᵉ    )
                           renaming ( trans to transᵉ  ) using ()

-- Assign handles to 3 modules of the Agda Standard Library.
import Function.Definitions                as FD
import Relation.Binary.PropositionalEquality as ≡
import Relation.Binary.Reasoning.Setoid    as SetoidReasoning

private variable
  α ρᵃ β ρᵇ γ ρᶜ δ ρᵈ ρ χ ℓ : Level
  Γ Δ : Type χ
  f : fst S
```

Note that the above imports include some adjustments to "standard Agda" syntax to suit our own taste. In particular, the following conventions used throughout the agda-algebras library and this paper: we use Type in place of Set, the infix long arrow symbol, _⟶_, instead of Func (the type of "setoid functions" discussed in §2.3 below), and the symbol _⟨$⟩_ in place of f (application of the map of a setoid function); we use fst and snd, and sometimes |_| and ‖_‖, to denote the first and second projections out of the product type _×_.

```
module _ {A : Type α }{B : A → Type β} where
  |_| : Σ[ x ∈ A ] B x → A
  |_| = fst
  ‖_‖ : (z : Σ[ a ∈ A ] B a) → B | z |
  ‖_‖ = snd
```

2.2 Setoids

A *setoid* is a pair $(A, \approx)$ where A is a type and $\approx$ is an equivalence relation on A. Setoids seem to have gotten a bad wrap in some parts of the interactive theorem proving community because of the extra overhead they require. However, we feel they are ideally suited to representing the basic objects of informal mathematics (i.e., sets) in a constructive, type-theoretic way.

In informal mathematical discourse, a set typically comes equipped with an equivalence relation manifesting the notion of equality of elements of the set. We often take this equivalence for granted or view it as self-evident; rarely do we take pains to define it explicitly. While well-suited to informal mathematics, this approach is inadequate for formal, machine-checked proofs.

The agda-algebras library was first developed without setoids, relying exclusively on the inductive equality type `_≡_`, defined in Agda.Builtin.Equality, along with some experimental, domain-specific types for equivalence classes, quotients, etc. One consequence of this design decision was that the formalization of many theorems required postulating function extensionality, an axiom that is known to be neither provable nor refutable in pure Martin-Löf type theory.[5]

In contrast, our current approach using setoids makes the equality relation of a given type explicit. A primary motivation for this choice is to avoid the need for additional axioms and to make it clearer that the formal proofs in the agda-algebras library are fully *constructive* (as defined in [13]) and confined to *Martin-Löf dependent type theory* (as defined in [14]). In particular, we make no appeals to classical axioms like Choice or Excluded Middle, nor do we postulate function extensionality at any point in the present work.[6] We are confident that the agda-algebras library is now fully constructive and free from any hidden assumptions or inconsistencies that could be used to fool a type-checker.[7]

2.3 Setoid functions

In addition to the Setoid type, much of our code employs the standard library's Func type which represents a function from one setoid to another and packages such a function with a proof (called cong) that the function respects the underlying setoid equalities. As mentioned above, we renamed Func to the more visually appealing infix long arrow symbol, `_⟶_`, and throughout the paper we refer to inhabitants of this type as "setoid functions."

An example of a setoid function is the identity function from a setoid to itself. We define it, along with a binary composition operation for setoid functions, `⟨∘⟩`, as follows.

```
id : {A : Setoid α ρᵃ} → A ⟶ A
id {A} = record { f = id ; cong = id }

_⟨∘⟩_ : {A : Setoid α ρᵃ} {B : Setoid β ρᵇ} {C : Setoid γ ρᶜ}
  →     B ⟶ C → A ⟶ B → A ⟶ C

f ⟨∘⟩ g = record { f = (_⟨$⟩_ f) ∘ (_⟨$⟩_ g)
                ; cong = (cong f) ∘ (cong g) }
```

[5] See the section *Function extensionality from univalence* in [8, 9].

[6] The *function extensionality axiom* asserts that two point-wise equal functions are equal. There remain some modules in the agda-algebras library that occasionally postulate this axiom, but we don't make use of the axiom here.

[7] As of 26 Nov 2021, the latest version of agda-algebras is 2.0.0; see [7].

Inverses of setoid functions

We begin by defining an inductive type that represents the *image* of a function.[8]

```
module _ {A : Setoid α ρᵃ}{B : Setoid β ρᵇ} where
  open Setoid B using ( _≈_ ; sym ) renaming ( Carrier to B )

  data Image_∋_ (f : A ⟶ B) : B → Type (α ⊔ β ⊔ ρᵇ) where
    eq : {b : B} → ∀ a → b ≈ f ⟨$⟩ a → Image f ∋ b
```

An inhabitant of `Image f ∋ b` is a dependent pair `(a , p)`, where `a : A` and `p : b ≈ f a` is a proof that `f` maps `a` to `b`. Since the proof that `b` belongs to the image of `f` is always accompanied by a witness `a : A`, we can actually *compute* a range-restricted right-inverse of `f`, as follows.

```
Inv : (f : A ⟶ B){b : B} → Image f ∋ b → Carrier A
Inv _ (eq a _) = a
```

For each `b : B`, given a pair `(a , p) : Image f ∋ b` witnessing the fact that `b` belongs to the image of `f`, the function `Inv` simply returns the witness `a`, which is a preimage of `b` under `f`. Let's formally verify that `Inv f` is indeed the (range-restricted) right-inverse of `f`.

```
InvIsInverseʳ : {f : A ⟶ B}{b : B}(q : Image f ∋ b) → f ⟨$⟩ (Inv f q) ≈ b
InvIsInverseʳ (eq _ p) = sym p
```

Injective and surjective setoid functions

If `f` is a setoid function from `(A, ≈ᴬ)` to `(B, ≈ᴮ)`, then we call `f` *injective* provided `∀ (a₀ a₁ : A), f ⟨$⟩ a₀ ≈ᴮ f ⟨$⟩ a₁` implies `a₀ ≈ᴬ a₁`; we call `f` *surjective* provided `∀ (b : B), ∃ (a : A)` such that `f ⟨$⟩ a ≈ᴮ b`. The Agda Standard Library represents injective functions on bare types by the type `Injective`, and uses this to define the `IsInjective` type to represent the property of being an injective setoid function. Similarly, the type `IsSurjective` represents the property of being a surjective setoid function. `SurjInv` represents the *right-inverse* of a surjective function. We reproduce the definitions and prove some of their properties inside the next submodule where we first set the stage by declaring two setoids `A` and `B`, naming their equality relations, and making some definitions from the standard library available.

```
module _ {A : Setoid α ρᵃ}{B : Setoid β ρᵇ} where
  open Setoid A using () renaming ( _≈_ to _≈ᴬ_ )
  open Setoid B using () renaming ( _≈_ to _≈ᴮ_ )
  open FD _≈ᴬ_ _≈ᴮ_

  IsInjective : (A ⟶ B) → Type (α ⊔ ρᵃ ⊔ ρᵇ)
  IsInjective f = Injective (_⟨$⟩_ f)

  IsSurjective : (A ⟶ B) → Type (α ⊔ β ⊔ ρᵇ)
  IsSurjective F = ∀ {y} → Image F ∋ y

  SurjInv : (f : A ⟶ B) → IsSurjective f → Carrier B → Carrier A
  SurjInv f fonto b = Inv f (fonto {b})
```

[8] cf. the Overture.Func.Inverses module of the agda-algebras library.

Proving that the composition of injective setoid functions is again injective is simply a matter of composing the two assumed witnesses to injectivity. Proving that surjectivity is preserved under composition is only slightly more involved.

```
module _ {𝐀 : Setoid α ρᵃ}{𝐁 : Setoid β ρᵇ}{𝐂 : Setoid γ ρᶜ}
         (f : 𝐀 ⟶ 𝐁)(g : 𝐁 ⟶ 𝐂) where

  ∘-IsInjective : IsInjective f → IsInjective g → IsInjective (g ⟨∘⟩ f)
  ∘-IsInjective finj ginj = finj ∘ ginj

  ∘-IsSurjective : IsSurjective f → IsSurjective g → IsSurjective (g ⟨∘⟩ f)
  ∘-IsSurjective fonto gonto {y} = Goal
    where
    mp : Image g ∋ y → Image g ⟨∘⟩ f ∋ y
    mp (eq c p) = η fonto
      where
      open Setoid 𝐂 using ( trans )
      η : Image f ∋ c → Image g ⟨∘⟩ f ∋ y
      η (eq a q) = eq a (trans p (cong g q))

    Goal : Image g ⟨∘⟩ f ∋ y
    Goal = mp gonto
```

Kernels of setoid functions

The *kernel* of a function f : A → B (where A and B are bare types) is defined informally by {(x , y) ∈ A × A : f x = f y }. This can be represented in Agda in a number of ways, but for our purposes it is most convenient to define the kernel as an inhabitant of a (unary) predicate over the square of the function's domain, as follows.

```
kernel : {A : Type α}{B : Type β} → Rel B ρ → (A → B) → Pred (A × A) ρ
kernel _≈_ f (x , y) = f x ≈ f y
```

The kernel of a *setoid* function f : $A \longrightarrow B$ is {(x , y) ∈ A × A : f ⟨$⟩ x ≈ f ⟨$⟩ y }, where _≈_ denotes equality in B. This can be formalized in Agda as follows.

```
module _ {A : Setoid α ρᵃ}{B : Setoid β ρᵇ} where
  open Setoid A using () renaming ( Carrier to A )

ker : (A ⟶ B) → Pred (A × A) ρᵇ
ker g (x , y) = g ⟨$⟩ x ≈ g ⟨$⟩ y where open Setoid B using ( _≈_ )
```

3 Types for Basic Universal Algebra

In this section we develop a working vocabulary and formal types for classical, single-sorted, set-based universal algebra. We cover a number of important concepts, but we limit ourselves to those concepts required in our formal proof of Birkhoff's HSP theorem. In each case, we give a type-theoretic version of the informal definition, followed by a formal implementation of the definition in MLTT using the Agda language.

This section is organized into the following subsections: §3.1 defines a general notion of *signature* of a structure and then defines a type that represent signatures; §3.2 does the same for *algebraic structures* and *product algebras*; §3.3 defines *homomorphisms*, *monomorphisms*, and *epimorphisms*, presents types that codify these concepts and formally verifies some of their basic properties; §§3.4–3.5 do the same for *subalgebras* and *terms*, respectively.

3.1 Signatures

In model theory, the *signature* of a structure is a quadruple $S = (\mathsf{C}, \mathsf{F}, \mathsf{R}, \rho)$ consisting of three (possibly empty) sets C, F, and R—called *constant*, *function*, and *relation* symbols, respectively—along with a function $\rho : \mathsf{C} + \mathsf{F} + \mathsf{R} \to \mathsf{N}$ that assigns an *arity* to each symbol. Often, but not always, N is taken to be the set of natural numbers.

As our focus here is universal algebra, we consider the restricted notion of an *algebraic signature*, that is, a signature for "purely algebraic" structures. Such a signature is a pair $S = (\mathsf{F}\ ,\ \rho)$ where F is a collection of *operation symbols* and $\rho : \mathsf{F} \to \mathsf{N}$ is an *arity function* which maps each operation symbol to its arity. Here, N denotes the *arity type*. Heuristically, the arity $\rho\ \mathsf{f}$ of an operation symbol $\mathsf{f} \in \mathsf{F}$ may be thought of as the number of arguments that f takes as "input."

The agda-algebras library represents an algebraic signature as an inhabitant of the following dependent pair type:

```
Signature : (𝓞 𝓥 : Level) → Type (lsuc (𝓞 ⊔ 𝓥))
Signature 𝓞 𝓥 = Σ[ F ∈ Type 𝓞 ] (F → Type 𝓥)
```

Using special syntax for the first and second projections—$\vert\_\vert$ and $\Vert\_\Vert$ (resp.)—if S : Signature $\mathcal{O}$ $\mathcal{V}$ is a signature, then $\vert\ S\ \vert$ denotes the set of operation symbols and $\Vert\ S\ \Vert$ denotes the arity function. Thus, if $\mathsf{f} : \vert\ S\ \vert$ is an operation symbol in the signature S, then $\Vert\ S\ \Vert\ \mathsf{f}$ is the arity of f.

We need to augment the ordinary Signature type so that it supports algebras over setoid domains. To do so—following Andreas Abel's lead (cf. [1])—we define an operator that translates an ordinary signature into a *setoid signature*, that is, a signature over a setoid domain. This raises a minor technical issue concerning the dependent types involved in the definition. Some readers might find the resolution of this issue instructive, so let's discuss it briefly. If we are given two operations f and g, a tuple $\mathsf{u} : \Vert\ S\ \Vert\ \mathsf{f} \to \mathsf{A}$ of arguments for f, and a tuple $\mathsf{v} : \Vert\ S\ \Vert\ \mathsf{g} \to \mathsf{A}$ of arguments for g, and if we know that f is identically equal to g—that is, $\mathsf{f} \equiv \mathsf{g}$ (intensionally)—then we should be able to check whether u and v are pointwise equal. Technically, though, u and v inhabit different types, so, in order to compare them, we must convince Agda that u and v inhabit the same type. Of course, this requires an appeal to the hypothesis $\mathsf{f} \equiv \mathsf{g}$, as we see in the definition of EqArgs below (adapted from Andreas Abel's development [1]), which neatly resolves this minor technicality.

```
EqArgs : {S : Signature 𝓞 𝓥}{ξ : Setoid α ρᵃ}
  →      ∀ {f g} → f ≡ g → (∥ S ∥ f → Carrier ξ) → (∥ S ∥ g → Carrier ξ) → Type (𝓥 ⊔ ρᵃ)

EqArgs {ξ = ξ} ≡.refl u v = ∀ i → u i ≈ v i where open Setoid ξ using ( _≈_ )
```

Finally, we are ready to define an operator which translates an ordinary (algebraic) signature into a signature of algebras over setoids. We denote this operator by $\langle\_\rangle$ and define it as follows.

```
⟨_⟩ : Signature 𝓞 𝓥 → Setoid α ρᵃ → Setoid _ _

Carrier (⟨ S ⟩ ξ)              = Σ[ f ∈ ∣ S ∣ ] (∥ S ∥ f → ξ .Carrier)
_≈ˢ_ (⟨ S ⟩ ξ)(f , u)(g , v) = Σ[ eqv ∈ f ≡ g ] EqArgs{ξ = ξ} eqv u v

reflᵉ  (isEquivalence (⟨ S ⟩ ξ))                          = ≡.refl , λ i → reflˢ  ξ
symᵉ   (isEquivalence (⟨ S ⟩ ξ)) (≡.refl , g)             = ≡.refl , λ i → symˢ   ξ (g i)
transᵉ (isEquivalence (⟨ S ⟩ ξ)) (≡.refl , g)(≡.refl , h) = ≡.refl , λ i → transˢ ξ (g i) (h i)
```

3.2 Algebras

Informally, an *algebraic structure in the signature* $S = (\mathsf{F}, \rho)$, or *S-algebra*, is denoted by $\mathbf{A} = (\mathsf{A}, \mathsf{F}^A)$ and consists of

- a *nonempty* set (or type) A, called the *domain* (or *carrier* or *universe*) of the algebra;
- a collection $\mathsf{F}^A := \{ \mathsf{f}^A \mid \mathsf{f} \in \mathsf{F}, \mathsf{f}^A : (\rho\ \mathsf{f} \to \mathsf{A}) \to \mathsf{A} \}$ of *operations* on A;
- a (potentially empty) collection of *identities* satisfied by elements and operations of $\mathbf{A}$.

The agda-algebras library represents algebras as inhabitants of a record type with two fields:[9]

- Domain, representing the domain of the algebra;
- Interp, representing the *interpretation* in the algebra of each operation symbol in S.

The Domain is a setoid whose Carrier denotes the domain of the algebra and whose equivalence relation denotes equality of elements of the domain.

Here is the definition of the Algebra type followed by an explanation of how the standard library's Func type is used to represent the interpretation of operation symbols in an algebra.

```
record Algebra α ρ : Type (𝓞 ⊔ 𝓥 ⊔ lsuc (α ⊔ ρ)) where
  field Domain : Setoid α ρ
        Interp : ⟨ S ⟩ Domain ⟶ Domain
```

Recall, we renamed Agda's Func type, preferring instead the long-arrow symbol ⟶, so the Interp field has type Func (⟨ S ⟩ Domain) Domain, a record type with two fields:

- a function f : Carrier (⟨ S ⟩ Domain) → Carrier Domain representing the operation;
- a proof cong : f Preserves _≈₁_ ⟶ _≈₂_ that the operation preserves the relevant setoid equalities.

Thus, for each operation symbol in the signature S, we have a setoid function f—with domain a power of Domain and codomain Domain—along with a proof that this function respects the setoid equalities. The latter means that the operation f is accompanied by a proof of the following: ∀ u v in Carrier (⟨ S ⟩ Domain), if u ≈₁ v, then f ⟨$⟩ u ≈₂ f ⟨$⟩ v.

In the agda-algebras library is defined some syntactic sugar that helps to make our formalizations easier to read and comprehend. The following are three examples of such syntax that we use below: if $\mathbf{A}$ is an algebra, then

- 𝔻[**A**] denotes the setoid Domain **A**,
- 𝕌[**A**] is the underlying carrier of the algebra **A**, and
- f ̂ **A** denotes the interpretation in the algebra **A** of the operation symbol f.

```
open Algebra
𝔻[_] : Algebra α ρᵃ → Setoid α ρᵃ
𝔻[ A ] = Domain A
𝕌[_] : Algebra α ρᵃ → Type α
𝕌[ A ] = Carrier (Domain A)
_̂_ : (f : ∣ S ∣)(A : Algebra α ρᵃ) → (∥ S ∥ f → 𝕌[ A ]) → 𝕌[ A ]
f ̂ A = λ a → (Interp A) ⟨$⟩ (f , a)
```

Universe levels of algebra types

The hierarchy of type universes in Agda is structured as follows: Type ℓ : Type (lsuc ℓ), Type (lsuc ℓ) : Type (lsuc (lsuc ℓ)), ... This means that Type ℓ has type Type (lsuc ℓ), etc. However, this does *not* imply that Type ℓ : Type (lsuc (lsuc ℓ)). In other words, Agda's

[9] We postpone introducing identities until §4.

universe hierarchy is *noncumulative*. This can be advantageous as it becomes possible to treat universe levels more generally and precisely. On the other hand, an unfortunate side-effect of this noncumulativity is that it can sometimes seem unreasonably difficult to convince Agda that a program or proof is correct. This aspect of the language was one of the few stumbling blocks we encountered while learning how to use Agda for formalizing universal algebra in type theory. Although some may consider this to be one of the least interesting and most technical aspects of this paper, others might find the presentation more helpful if we resist the urge to gloss over these technicalities. Therefore, it seems worthwhile to explain how we make use of the general universe lifting and lowering functions, available in the Agda Standard Library, to develop domain-specific tools for dealing with Agda's noncumulative universe hierarchy.

Let us be more concrete about what is at issue by considering a typical example. Agda frequently encounters problems during the type-checking process and responds by printing a message like the following.

```
HSP.lagda:498,20-23
α != 𝓞 ⊔ 𝓥 ⊔ (lsuc α) when checking that... has type...
```

Here Agda informs us that it encountered universe level α on line 498 of the HSP module, where it was expecting level $\mathcal{O} \sqcup \mathcal{V} \sqcup (\mathsf{lsuc}\ \alpha)$. In this case, we tried to use an algebra inhabiting the type Algebra $\alpha\ \rho^a$ whereas Agda expected an inhabitant of the type Algebra $(\mathcal{O} \sqcup \mathcal{V} \sqcup (\mathsf{lsuc}\ \alpha))\ \rho^a$. To resolve such problems, we use the Lift record type of the Agda Standard Library, which takes a type inhabiting a particular universe and embeds it into a higher universe. Specializing the Lift type to our domain of interest, the agda-algebras library defines a function called Lift-Alg.

```
module _ (𝐀 : Algebra α ρᵃ) where
  open Setoid 𝔻[ 𝐀 ] using ( _≈_ ; refl ; sym ; trans ) ; open Level
  Lift-Algˡ : (ℓ : Level) → Algebra (α ⊔ ℓ) ρᵃ
  Domain (Lift-Algˡ ℓ) =
    record { Carrier        = Lift ℓ 𝕌[ 𝐀 ]
           ; _≈_            = λ x y → lower x ≈ lower y
           ; isEquivalence = record { refl = refl ; sym = sym ; trans = trans }}

  Interp (Lift-Algˡ ℓ) ⟨$⟩ (f , la) = lift ((f ̂ 𝐀) (lower ∘ la))
  cong (Interp (Lift-Algˡ ℓ)) (≡.refl , lab) = cong (Interp 𝐀) ((≡.refl , lab))

  Lift-Algʳ : (ℓ : Level) → Algebra α (ρᵃ ⊔ ℓ)
  Domain (Lift-Algʳ ℓ) =
    record { Carrier        = 𝕌[ 𝐀 ]
           ; _≈_            = λ x y → Lift ℓ (x ≈ y)
           ; isEquivalence = record { refl = lift refl
                                    ; sym = lift ∘ sym ∘ lower
                                    ; trans = λ x y → lift (trans (lower x)(lower y)) }}

  Interp (Lift-Algʳ ℓ ) ⟨$⟩ (f , la) = (f ̂ 𝐀) la
  cong (Interp (Lift-Algʳ ℓ))(≡.refl , lab) = lift(cong(Interp 𝐀)(≡.refl , λ i → lower (lab i)))

Lift-Alg : (𝐀 : Algebra α ρᵃ)(ℓ₀ ℓ₁ : Level) → Algebra (α ⊔ ℓ₀) (ρᵃ ⊔ ℓ₁)
Lift-Alg 𝐀 ℓ₀ ℓ₁ = Lift-Algʳ (Lift-Algˡ 𝐀 ℓ₀) ℓ₁
```

To see why the Lift-Alg function is useful, recall that our definition of the algebra record type uses two universe level parameters corresponding to those of the algebra's underlying domain

setoid. Concretely, an algebra of type Algebra α ρ^a has a Domain of type Setoid α ρ^a. This packages a "carrier set" (Carrier), inhabiting Type α, with an equality on Carrier of type Rel Carrier ρ^a. The Lift-Alg function takes an algebra—one whose carrier inhabits Type α with equality of type Rel Carrier ρ^a—and constructs a new algebra whose carrier inhabits Type (α $\sqcup$ ℓ_0) with equality of type Rel Carrier (ρ^a $\sqcup$ ℓ_1). This lifting operation would be worthless without a useful semantic connection between the input and output algebras. Fortunately, it is easy to prove that Lift-Alg is an *algebraic invariant*, which is to say that the resulting "lifted" algebra has the same algebraic properties as the original algebra, a fact we will codify later in a type called Lift-≅.

Product Algebras

Here we review the (informal) definition of the *product* of a family of S-algebras and then define a type which formalizes this notion in Agda. Let ι be a universe and I : Type ι a type (the "indexing type"). Then the dependent function type $\mathcal{A}$: I → Algebra α ρ^a represents an *indexed family of algebras*. Denote by $\bigsqcap$ $\mathcal{A}$ the *product of algebras* in $\mathcal{A}$ (or *product algebra*), by which we mean the algebra whose domain is the Cartesian product Π i : I , 𝔻[$\mathcal{A}$ i] of the domains of the algebras in $\mathcal{A}$, and whose operations are those arising by pointwise interpretation in the obvious way: if f is a J-ary operation symbol and if a : Π i : I , J → 𝔻[$\mathcal{A}$ i] is, for each i : I, a J-tuple of elements of the domain 𝔻[$\mathcal{A}$ i], then we define the interpretation of f in $\bigsqcap$ $\mathcal{A}$ by

(f ̂ $\bigsqcap$ $\mathcal{A}$) a := λ (i : I) → (f ̂ $\mathcal{A}$ i)(a i).

In the agda-algebras library we define the function $\bigsqcap$ which formalizes this notion of *product algebra* in MLTT. Here is the formal definition.

```
module _ {ι : Level}{I : Type ι } where
  ⨅ : (𝒜 : I → Algebra α ρᵃ) → Algebra (α ⊔ ι) (ρᵃ ⊔ ι)
Domain (⨅ 𝒜) =
  record { Carrier = ∀ i → 𝕌[ 𝒜 i ]
         ; _≈_ = λ a b → ∀ i → (_≈ˢ_ 𝔻[ 𝒜 i ]) (a i)(b i)
         ; isEquivalence =
             record { refl  = λ i →     reflᵉ  (isEquivalence 𝔻[ 𝒜 i ])
                    ; sym   = λ x i →   symᵉ    (isEquivalence 𝔻[ 𝒜 i ])(x i)
                    ; trans = λ x y i → transᵉ  (isEquivalence 𝔻[ 𝒜 i ])(x i)(y i) }}
Interp (⨅ 𝒜) ⟨$⟩ (f , a) = λ i → (f ̂ (𝒜 i)) (flip a i)
cong (Interp (⨅ 𝒜)) (≡.refl , f=g ) = λ i → cong (Interp (𝒜 i)) (≡.refl , flip f=g i )
```

3.3 Homomorphisms

Throughout this section, and the rest of the paper unless stated otherwise, $\mathbf{A}$ and $\mathbf{B}$ will denote S-algebras inhabiting the types Algebra α ρ^a and Algebra β ρ^b, respectively.

A *homomorphism* (or "hom") from $\mathbf{A}$ to $\mathbf{B}$ is a setoid function h : 𝔻[$\mathbf{A}$] ⟶ 𝔻[$\mathbf{B}$] that is *compatible* with all basic operations; that is, for every operation symbol f : ∣ S ∣ and all tuples a : ∥ S ∥ f → 𝕌[$\mathbf{A}$], we have h ⟨$⟩ (f ̂ $\mathbf{A}$) a ≈ (f ̂ $\mathbf{B}$) h ⟨$⟩ (a _). To formalize this concept in Agda, we first define the type compatible-map-op representing the assertion that a given setoid function h : 𝔻[$\mathbf{A}$] ⟶ 𝔻[$\mathbf{B}$] commutes with a given operation symbol f. Then we generalize over operation symbols in the definition of compatible-map, the type of compatible maps from (the domain of) A to (the domain of) $\mathbf{B}$.

```
module _ (𝐀 : Algebra α ρᵃ)(𝐁 : Algebra β ρᵇ) where

compatible-map-op : (𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) → ∣ S ∣ → Type _
```

```
compatible-map-op h f = ∀ {a} → h ⟨$⟩ (f ̂ 𝐀) a ≈ (f ̂ 𝐁) λ x → h ⟨$⟩ (a x)
  where open Setoid 𝔻[ 𝐁 ] using ( _≈_ )
compatible-map : (𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) → Type _
compatible-map h = ∀ {f} → compatible-map-op h f
```

Using these we define a record type IsHom representing the property of being a homomorphism, and finally the type hom of homomorphisms from $\mathbf{A}$ to B.

```
record IsHom (h : 𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) : Type (𝓞 ⊔ 𝓥 ⊔ α ⊔ ρᵇ) where
  constructor mkhom ; field compatible : compatible-map h
hom : Type _
hom = Σ (𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) IsHom
```

Thus, an inhabitant of hom is a pair (h , p) whose first component is a setoid function from the domain of $\mathbf{A}$ to that of $\mathbf{B}$ and whose second component is p : IsHom h, a proof that h is a homomorphism.

A *monomorphism* (resp. *epimorphism*) is an injective (resp. surjective) homomorphism. The agda-algebras library defines types IsMon and IsEpi to represent these properties, as well as mon and epi, the types of monomorphisms and epimorphisms, respectively.

```
record IsMon (h : 𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) : Type (𝓞 ⊔ 𝓥 ⊔ α ⊔ ρᵃ ⊔ ρᵇ) where
  field isHom : IsHom h
        isInjective : IsInjective h

  HomReduct : hom
  HomReduct = h , isHom

mon : Type _
mon = Σ (𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) IsMon
```

As with hom, the type mon is a dependent product type; each inhabitant is a pair consisting of a setoid function, say, h, along with a proof that h is a monomorphism.

```
record IsEpi (h : 𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) : Type (𝓞 ⊔ 𝓥 ⊔ α ⊔ β ⊔ ρᵇ) where
  field isHom : IsHom h
        isSurjective : IsSurjective h

  HomReduct : hom
  HomReduct = h , isHom

epi : Type _
epi = Σ (𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]) IsEpi
Here are two mere utilities that are useful for translating between types.
open IsHom ; open IsMon ; open IsEpi

module _ (𝐀 : Algebra α ρᵃ)(𝐁 : Algebra β ρᵇ) where

  mon→intohom : mon 𝐀 𝐁 → Σ[ h ∈ hom 𝐀 𝐁 ] IsInjective ∣ h ∣
  mon→intohom (hh , hhM) = (hh , isHom hhM) , isInjective hhM

  epi→ontohom : epi 𝐀 𝐁 → Σ[ h ∈ hom 𝐀 𝐁 ] IsSurjective ∣ h ∣
  epi→ontohom (hh , hhE) = (hh , isHom hhE) , isSurjective hhE
```

Composition of homomorphisms

The composition of homomorphisms is again a homomorphism, and similarly for epimorphisms (and monomorphisms).

```
module _ {𝐀 : Algebra α ρᵃ} {𝐁 : Algebra β ρᵇ} {𝐂 : Algebra γ ρᶜ}
         {g : 𝔻[ 𝐀 ] ⟶ 𝔻[ 𝐁 ]}{h : 𝔻[ 𝐁 ] ⟶ 𝔻[ 𝐂 ]} where

  open Setoid 𝔻[ 𝐂 ] using ( trans )

  ∘-is-hom : IsHom 𝐀 𝐁 g → IsHom 𝐁 𝐂 h → IsHom 𝐀 𝐂 (h ⟨∘⟩ g)
  ∘-is-hom ghom hhom = mkhom c
    where
    c : compatible-map 𝐀 𝐂 (h ⟨∘⟩ g)
    c = trans (cong h (compatible ghom)) (compatible hhom)

  ∘-is-epi : IsEpi 𝐀 𝐁 g → IsEpi 𝐁 𝐂 h → IsEpi 𝐀 𝐂 (h ⟨∘⟩ g)
  ∘-is-epi gE hE = record { isHom = ∘-is-hom (isHom gE) (isHom hE)
                          ; isSurjective = ∘-IsSurjective g h (isSurjective gE) (isSurjective hE) }

module _ {𝐀 : Algebra α ρᵃ} {𝐁 : Algebra β ρᵇ} {𝐂 : Algebra γ ρᶜ} where

  ∘-hom : hom 𝐀 𝐁 → hom 𝐁 𝐂 → hom 𝐀 𝐂
  ∘-hom (h , hhom) (g , ghom) = (g ⟨∘⟩ h) , ∘-is-hom hhom ghom

  ∘-epi : epi 𝐀 𝐁 → epi 𝐁 𝐂      → epi 𝐀 𝐂
  ∘-epi (h , hepi) (g , gepi) = (g ⟨∘⟩ h) , ∘-is-epi hepi gepi
```

Universe lifting of homomorphisms

Here we define the identity homomorphism for setoid algebras. Then we prove that the operations of lifting and lowering of a setoid algebra are homomorphisms.

```
𝑖𝑑 : {𝐀 : Algebra α ρᵃ} → hom 𝐀 𝐀
𝑖𝑑 {𝐀 = 𝐀} = 𝑖𝑑 , mkhom (reflexive ≡.refl) where open Setoid ( Domain 𝐀 ) using ( reflexive )

module _ {𝐀 : Algebra α ρᵃ}{ℓ : Level} where
  open Setoid 𝔻[ 𝐀 ]              using ( reflexive ) renaming ( _≈_ to _≈₁_ ; refl to refl₁ )
  open Setoid 𝔻[ Lift-Algˡ 𝐀 ℓ ] using ()            renaming ( _≈_ to _≈ˡ_ ; refl to reflˡ )
  open Setoid 𝔻[ Lift-Algʳ 𝐀 ℓ ] using ()            renaming ( _≈_ to _≈ʳ_ ; refl to reflʳ )
  open Level

  ToLiftˡ : hom 𝐀 (Lift-Algˡ 𝐀 ℓ)
  ToLiftˡ = record { f = lift ; cong = id } , mkhom (reflexive ≡.refl)

  FromLiftˡ : hom (Lift-Algˡ 𝐀 ℓ) 𝐀
  FromLiftˡ = record { f = lower ; cong = id } , mkhom reflˡ

  ToFromLiftˡ : ∀ b → | ToLiftˡ | ⟨$⟩ (| FromLiftˡ | ⟨$⟩ b) ≈ˡ b
  ToFromLiftˡ b = refl₁

  FromToLiftˡ : ∀ a → | FromLiftˡ | ⟨$⟩ (| ToLiftˡ | ⟨$⟩ a) ≈₁ a
  FromToLiftˡ a = refl₁

  ToLiftʳ : hom 𝐀 (Lift-Algʳ 𝐀 ℓ)
  ToLiftʳ = record { f = id ; cong = lift } , mkhom (lift (reflexive ≡.refl))

  FromLiftʳ : hom (Lift-Algʳ 𝐀 ℓ) 𝐀
  FromLiftʳ = record { f = id ; cong = lower } , mkhom reflˡ

  ToFromLiftʳ : ∀ b → | ToLiftʳ | ⟨$⟩ (| FromLiftʳ | ⟨$⟩ b) ≈ʳ b
```

```
  ToFromLift^r b = lift refl₁

  FromToLift^r : ∀ a → | FromLift^r | ⟨$⟩ (| ToLift^r | ⟨$⟩ a) ≈₁ a
  FromToLift^r a = refl₁


module _ {𝐀 : Algebra α ρ^a}{ℓ r : Level} where
  open Setoid 𝔻[ 𝐀 ]              using ( refl )
  open Setoid 𝔻[ Lift-Alg 𝐀 ℓ r ] using ( _≈_ )
  open Level

  ToLift : hom 𝐀 (Lift-Alg 𝐀 ℓ r)
  ToLift = ∘-hom ToLift^l ToLift^r

  FromLift : hom (Lift-Alg 𝐀 ℓ r) 𝐀
  FromLift = ∘-hom FromLift^r FromLift^l

  ToFromLift : ∀ b → | ToLift | ⟨$⟩ (| FromLift | ⟨$⟩ b) ≈ b
  ToFromLift b = lift refl

  ToLift-epi : epi 𝐀 (Lift-Alg 𝐀 ℓ r)
  ToLift-epi = | ToLift | , record { isHom = ‖ ToLift ‖
                                   ; isSurjective = λ {y} → eq (| FromLift | ⟨$⟩ y) (ToFromLift y) }
```

Homomorphisms of product algebras

Suppose we have an algebra $\mathbf{A}$, a type $\mathsf{I} : \mathsf{Type}\ \mathcal{I}$, and a family $\mathcal{B} : \mathsf{I} \to \mathsf{Algebra}\ \beta\ \rho^b$ of algebras. We sometimes refer to the inhabitants of I as *indices*, and call $\mathcal{B}$ an *indexed family of algebras*. If in addition we have a family $\mathcal{h} : (\mathsf{i} : \mathsf{I}) \to \mathsf{hom}\ \mathbf{A}\ (\mathcal{B}\ \mathsf{i})$ of homomorphisms, then we can construct a homomorphism from $\mathbf{A}$ to the product $\prod \mathcal{B}$ in the natural way. We codify the latter in dependent type theory as follows.

```
module _ {ι : Level}{I : Type ι}{𝐀 : Algebra α ρ^a}(ℬ : I → Algebra β ρ^b) where
  ⨅-hom-co : (∀(i : I) → hom 𝐀 (ℬ i)) → hom 𝐀 (⨅ ℬ)
  ⨅-hom-co h = h , hhom
    where
    h : 𝔻[ 𝐀 ] ⟶ 𝔻[ ⨅ ℬ ]
    h ⟨$⟩ a = λ i → | h i | ⟨$⟩ a
    cong h xy i = cong | h i | xy
    hhom : IsHom 𝐀 (⨅ ℬ) h
    compatible hhom = λ i → compatible ‖ h i ‖
```

Factorization of homomorphisms

Another basic fact about homomorphisms that we formalize in the agda-algebras library (as the type HomFactor) is the following factorization theorem: if $\mathsf{g} : \mathsf{hom}\ \mathbf{A}\ \mathbf{B}$, $\mathsf{h} : \mathsf{hom}\ \mathbf{A}\ \mathbf{C}$, h is surjective, and $\mathsf{ker}\ \mathsf{h} \subseteq \mathsf{ker}\ \mathsf{g}$, then there exists $\varphi : \mathsf{hom}\ \mathbf{C}\ \mathbf{B}$ such that $\mathsf{g} = \varphi \circ \mathsf{h}$.

```
module _ {𝐀 : Algebra α ρ^a}(𝐁 : Algebra β ρ^b){𝐂 : Algebra γ ρ^c}
         (gh : hom 𝐀 𝐁)(hh : hom 𝐀 𝐂) where
  open Setoid 𝔻[ 𝐁 ] using () renaming ( _≈_ to _≈₂_ )
  open Setoid 𝔻[ 𝐂 ] using () renaming ( _≈_ to _≈₃_ )
  private gfunc = | gh | ; g = _⟨$⟩_ gfunc ; hfunc = | hh | ; h = _⟨$⟩_ hfunc

  HomFactor :     kernel _≈₃_ h ⊆ kernel _≈₂_ g
```

```
    →              IsSurjective hfunc
    →              Σ[ φ ∈ hom 𝐂 𝐁 ] ∀ a → g a ≈₂ | φ | ⟨$⟩ h a
  HomFactor Khg hE = (φmap , φhom) , gφh
    where
    kerpres : ∀ a₀ a₁ → h a₀ ≈₃ h a₁ → g a₀ ≈₂ g a₁
    kerpres a₀ a₁ hyp = Khg hyp

    h⁻¹ : 𝕌[ 𝐂 ] → 𝕌[ 𝐀 ]
    h⁻¹ = SurjInv hfunc hE

    η : ∀ {c} → h (h⁻¹ c) ≈₃ c
    η = InvIsInverseʳ hE

    open Setoid 𝔻[ 𝐂 ] using ( sym ; trans )
    ζ : ∀{x y} → x ≈₃ y → h (h⁻¹ x) ≈₃ h (h⁻¹ y)
    ζ xy = trans η (trans xy (sym η))

    φmap : 𝔻[ 𝐂 ] ⟶ 𝔻[ 𝐁 ]
    _⟨$⟩_ φmap = g ∘ h⁻¹
    cong φmap = Khg ∘ ζ

    open _⟶_ φmap using () renaming (cong to φcong)

    gφh : (a : 𝕌[ 𝐀 ]) → g a ≈₂ φmap ⟨$⟩ h a
    gφh a = Khg (sym η)

    φcomp : compatible-map 𝐂 𝐁 φmap
    φcomp {f}{c} =
      begin
        φmap ⟨$⟩ (f ̂ 𝐂) c                ≈˘⟨ φcong (cong (Interp 𝐂) (≡.refl , λ _ → η)) ⟩
        g(h⁻¹(   (f ̂ 𝐂) (h ∘  h⁻¹ ∘ c ))) ≈˘⟨ φcong (compatible ∥ hh ∥)                  ⟩
        g(h⁻¹(h( (f ̂ 𝐀) (     h⁻¹ ∘ c )))) ≈˘⟨ gφh ((f ̂ 𝐀)(h⁻¹ ∘ c))                ⟩
        g(       (f ̂ 𝐀) (     h⁻¹ ∘ c ))  ≈⟨  compatible ∥ gh ∥                      ⟩
                 (f ̂ 𝐁) (g ∘ ( h⁻¹ ∘ c ))  ■ where open SetoidReasoning 𝔻[ 𝐁 ]

    φhom : IsHom 𝐂 𝐁 φmap
    compatible φhom = φcomp
```

Isomorphisms

Two structures are *isomorphic* provided there are homomorphisms from each to the other that compose to the identity. In the agda-algebras library we codify this notion as well as some of its obvious consequences, as a record type called _≅_. Note that the definition, shown below, includes a proof of the fact that the maps to and from are bijective, which makes this fact more accessible.

```
module _ (𝐀 : Algebra α ρᵃ) (𝐁 : Algebra β ρᵇ) where
  open Setoid 𝔻[ 𝐀 ] using () renaming ( _≈_ to _≈ᴬ_ )
  open Setoid 𝔻[ 𝐁 ] using () renaming ( _≈_ to _≈ᴮ_ )

record _≅_ : Type (𝓞 ⊔ 𝓥 ⊔ α ⊔ ρᵃ ⊔ β ⊔ ρᵇ ) where
  constructor mkiso
  field       to : hom 𝐀 𝐁
              from : hom 𝐁 𝐀
              to∼from : ∀ b → | to |   ⟨$⟩ (| from | ⟨$⟩ b) ≈ᴮ b
              from∼to : ∀ a → | from | ⟨$⟩ (| to |   ⟨$⟩ a) ≈ᴬ a
```

```
  toIsSurjective : IsSurjective | to |
  toIsSurjective {y} = eq (| from | ⟨$⟩ y) (sym (to∼from y))
    where open Setoid 𝔻[ 𝐁 ] using ( sym )

  toIsInjective : IsInjective | to |
  toIsInjective {x}{y} xy = trans (sym (from∼to x)) (trans ξ (from∼to y))
    where
    open Setoid 𝔻[ 𝐀 ] using ( sym ; trans )
    ξ : | from | ⟨$⟩ (| to | ⟨$⟩ x) ≈ᴬ | from | ⟨$⟩ (| to | ⟨$⟩ y)
    ξ = cong | from | xy

  fromIsSurjective : IsSurjective | from |
  fromIsSurjective {x} = eq (| to | ⟨$⟩ x) (sym (from∼to x))
    where open Setoid 𝔻[ 𝐀 ] using ( sym )

  fromIsInjective : IsInjective | from |
  fromIsInjective {x}{y} xy = trans (sym (to∼from x)) (trans ξ (to∼from y))
    where
    open Setoid 𝔻[ 𝐁 ] using ( sym ; trans )
    ξ : | to | ⟨$⟩ (| from | ⟨$⟩ x) ≈ᴮ | to | ⟨$⟩ (| from | ⟨$⟩ y)
    ξ = cong | to | xy

  open _≅_
```

It is easy to prove that _≅_ is an equivalence relation, as follows.

```
≅-refl : Reflexive (_≅_ {α}{ρᵃ})
≅-refl {α}{ρᵃ}{𝐀} = mkiso 𝑖𝑑 𝑖𝑑 (λ b → refl) λ a → refl where open Setoid 𝔻[ 𝐀 ] using ( refl )
≅-sym : Sym (_≅_{β}{ρᵇ}) (_≅_{α}{ρᵃ})
≅-sym φ = mkiso (from φ) (to φ) (from∼to φ) (to∼from φ)

≅-trans : Trans (_≅_ {α}{ρᵃ}) (_≅_{β}{ρᵇ}) (_≅_{α}{ρᵃ}{γ}{ρᶜ})
≅-trans {ρᶜ = ρᶜ}{𝐀}{𝐁}{𝐂} ab bc = mkiso f g τ ν
  where
    f : hom 𝐀 𝐂                ; g : hom 𝐂 𝐀
    f = ∘-hom (to ab) (to bc) ; g = ∘-hom (from bc) (from ab)

    open Setoid 𝔻[ 𝐀 ] using ( _≈_ ; trans )
    open Setoid 𝔻[ 𝐂 ] using () renaming ( _≈_ to _≈ᶜ_ ; trans to transᶜ )

    τ : ∀ b → | f | ⟨$⟩ (| g | ⟨$⟩ b) ≈ᶜ b
    τ b = transᶜ (cong | to bc | (to∼from ab (| from bc | ⟨$⟩ b))) (to∼from bc b)

    ν : ∀ a → | g | ⟨$⟩ (| f | ⟨$⟩ a) ≈ a
    ν a = trans (cong | from ab | (from∼to bc (| to ab | ⟨$⟩ a))) (from∼to ab a)
```

Lift-Alg is an algebraic invariant

The Lift-Alg operation neatly resolves the technical problem arising from the noncumulativity of Agda's universe hierarchy. It does so without changing the algebraic semantics because isomorphism classes of algebras are closed under Lift-Alg.

```
  module _ {𝐀 : Algebra α ρᵃ}{ℓ : Level} where
    Lift-≅ˡ : 𝐀 ≅ (Lift-Algˡ 𝐀 ℓ)
    Lift-≅ˡ = mkiso ToLiftˡ FromLiftˡ (ToFromLiftˡ{𝐀 = 𝐀}) (FromToLiftˡ{𝐀 = 𝐀}{ℓ})
```

```
  Lift-≅ʳ : 𝐀 ≅ (Lift-Algʳ 𝐀 ℓ)
  Lift-≅ʳ = mkiso ToLiftʳ FromLiftʳ (ToFromLiftʳ{𝐀 = 𝐀}) (FromToLiftʳ{𝐀 = 𝐀}{ℓ})

Lift-≅ : {𝐀 : Algebra α ρᵃ}{ℓ ρ : Level} → 𝐀 ≅ (Lift-Alg 𝐀 ℓ ρ)
Lift-≅ = ≅-trans Lift-≅ˡ Lift-≅ʳ
```

Homomorphic images

Here we describe what we have found to be the most useful way to represent the class of *homomorphic images* of an algebra in MLTT. For future reference, we also record the fact that an algebra is its own homomorphic image. (Here and in agda-algebras we use the shorthand ov $\alpha := \mathcal{O} \sqcup \mathcal{V} \sqcup \alpha$, for any level α.)

```
ov : Level → Level
ov α = 𝓞 ⊔ 𝓥 ⊔ lsuc α

_IsHomImageOf_ : (𝐁 : Algebra β ρᵇ)(𝐀 : Algebra α ρᵃ) → Type _
𝐁 IsHomImageOf 𝐀 = Σ[ φ ∈ hom 𝐀 𝐁 ] IsSurjective | φ |

HomImages : Algebra α ρᵃ → Type (α ⊔ ρᵃ ⊔ ov (β ⊔ ρᵇ))
HomImages {β = β}{ρᵇ = ρᵇ} 𝐀 = Σ[ 𝐁 ∈ Algebra β ρᵇ ] 𝐁 IsHomImageOf 𝐀

IdHomImage : {𝐀 : Algebra α ρᵃ} → 𝐀 IsHomImageOf 𝐀
IdHomImage {α = α}{𝐀 = 𝐀} = 𝑖𝑑 , λ {y} → Image_∋_.eq y refl
  where open Setoid 𝔻[ 𝐀 ] using ( refl )
```

These types should be self-explanatory, but just to be sure, we pause to describe the semantics of the Sigma type appearing in the definition of HomImages. If $\mathbf{A}$: Algebra α ρ^a is an S-algebra, then HomImages $\mathbf{A}$ denotes the type of pairs ($\mathbf{B}$, p) such that $\mathbf{B}$: Algebra β ρ^b and p is a proof that there exists a homomorphism from $\mathbf{A}$ onto $\mathbf{B}$.

3.4 Subalgebras

Given S-algebras $\mathbf{A}$ and $\mathbf{B}$, we say that $\mathbf{A}$ is a *subalgebra* of $\mathbf{A}$ and write $\mathbf{A} \leq \mathbf{B}$ just in case $\mathbf{A}$ can be *homomorphically embedded* in $\mathbf{B}$; in other terms, $\mathbf{A} \leq \mathbf{B}$ iff there exists an injective homomorphism from $\mathbf{A}$ to $\mathbf{B}$. The following definition codifies the *binary subalgebra relation*, _≤_, on the class of S-algebras.

```
_≤_ : Algebra α ρᵃ → Algebra β ρᵇ → Type _
𝐀 ≤ 𝐁 = Σ[ h ∈ hom 𝐀 𝐁 ] IsInjective | h |
```

Obviously the subalgebra relation is reflexive by the identity monomorphism; it is also transitive since composition of monomorphisms is a monomorphism.

```
  ≤-reflexive : {𝐀 : Algebra α ρᵃ} → 𝐀 ≤ 𝐀
  ≤-reflexive {𝐀 = 𝐀} = 𝑖𝑑 , id
  ≤-transitive : {𝐀 : Algebra α ρᵃ}{𝐁 : Algebra β ρᵇ}{𝐂 : Algebra γ ρᶜ}
    →            𝐀 ≤ 𝐁 → 𝐁 ≤ 𝐂 → 𝐀 ≤ 𝐂
  ≤-transitive ( f , finj ) ( g , ginj ) = (∘-hom f g ) , ∘-IsInjective | f | | g | finj ginj
```

If $\mathscr{A} : I \to$ Algebra α ρ^a, $\mathscr{B} : I \to$ Algebra β ρ^b (families of S-algebras) and $\mathscr{B}$ i ≤ $\mathscr{A}$ i for all i : I, then $\bigcap \mathscr{B}$ is a subalgebra of $\bigcap \mathscr{A}$.

```
  module _ {ι : Level} {I : Type ι}{𝒜 : I → Algebra α ρᵃ}{ℬ : I → Algebra β ρᵇ} where

  ⨅-≤ : (∀ i → ℬ i ≤ 𝒜 i) → ⨅ ℬ ≤ ⨅ 𝒜
```

```
⨅-≤ B≤A = (hfunc , hhom) , hM
  where
  hi : ∀ i → hom (𝒝 i) (𝒜 i)
  hi i = ∣ B≤A i ∣
  hfunc : 𝔻[ ⨅ 𝒝 ] ⟶ 𝔻[ ⨅ 𝒜 ]
  (hfunc ⟨$⟩ x) i = ∣ hi i ∣ ⟨$⟩ x i
  cong hfunc = λ xy i → cong ∣ hi i ∣ (xy i)
  hhom : IsHom (⨅ 𝒝) (⨅ 𝒜) hfunc
  compatible hhom = λ i → compatible ∥ hi i ∥
  hM : IsInjective hfunc
  hM = λ xy i → ∥ B≤A i ∥ (xy i)
```

We conclude this brief subsection on subalgebras with two easy facts that will be useful later. The first merely converts a monomorphism into a pair in the subalgebra relation while the second is an algebraic invariance property of $\leq$.

```
mon→≤  : {𝐀 : Algebra α ρᵃ}{𝐁 : Algebra β ρᵇ} → mon 𝐀 𝐁 → 𝐀 ≤ 𝐁
mon→≤ {𝐀 = 𝐀}{𝐁} x = mon→intohom 𝐀 𝐁 x

≅-trans-≤ : {𝐀 : Algebra α ρᵃ}{𝐁 : Algebra β ρᵇ}{𝐂 : Algebra γ ρᶜ}
  →          𝐀 ≅ 𝐁 → 𝐁 ≤ 𝐂 → 𝐀 ≤ 𝐂
≅-trans-≤ A≅B (h , hinj) = (∘-hom (to A≅B) h) , (∘-IsInjective ∣ to A≅B ∣ ∣ h ∣ (toIsInjective A≅B) hinj)
```

3.5 Terms

Fix a signature S and let X denote an arbitrary nonempty collection of variable symbols. Such a collection of variable symbols is called a *context*. Assume the symbols in X are distinct from the operation symbols of S, that is $X \cap \mid S \mid = \emptyset$. A *word* in the language of S is a finite sequence of members of $X \cup \mid S \mid$. We denote the concatenation of such sequences by simple juxtaposition. Let S_0 denote the set of nullary operation symbols of S. We define by induction on n the sets T_n of *words* over $X \cup \mid S \mid$ as follows (cf. [3, Def. 4.19]): $T_0 := X \cup S_0$ and $T_{n+1} := T_n \cup \mathcal{T}_n$, where $\mathcal{T}_n$ is the collection of all f t such that f : $\mid S \mid$ and t : $\parallel S \parallel$ f $\to T_n$. (Recall, $\parallel S \parallel$ f is the arity of the operation symbol f.) An *S-term* is a term in the language of S and the collection of all *S-terms* in the context X is given by Term X := $\bigcup_n T_n$.

As even its informal definition of Term X is recursive, it should come as no surprise that the semantics of terms can be usefully represented in type theory as an inductive type. Indeed, here is such a representation.

```
data Term (X : Type χ ) : Type (ov χ) where
  ℊ : X → Term X
  node : (f : ∣ S ∣)(t : ∥ S ∥ f → Term X) → Term X
```

This basic inductive type represents each term as a tree with an operation symbol at each node and a variable symbol at each leaf ℊ; hence the constructor names (ℊ for "generator" and node for "node").

The term algebra

We enrich the Term type with an inductive type _≃_ representing equality of terms, then we roll up into a setoid the types Term and _≃_ along with a proof that _≃_ is an equivalence

relation. Ultimately we use this setoid of S-terms as the domain of an algebra, called the *term algebra in the signature* S. Here is the equality type on terms.

```
module _ {X : Type χ } where

data _≃_ : Term X → Term X → Type (ov χ) where
  rfl : {x y : X} → x ≡ y → (ℊ x) ≃ (ℊ y)
  gnl : ∀ {f}{s t : ∥ S ∥ f → Term X} → (∀ i → (s i) ≃ (t i)) → (node f s) ≃ (node f t)
```

It's easy to show that this is an equivalence relation on terms, as follows.

```
≃-isRefl   : Reflexive     _≃_
≃-isRefl {ℊ _} = rfl ≡.refl
≃-isRefl {node _ _} = gnl (λ _ → ≃-isRefl)

≃-isSym   : Symmetric    _≃_
≃-isSym (rfl x) = rfl (≡.sym x)
≃-isSym (gnl x) = gnl (λ i → ≃-isSym (x i))

≃-isTrans : Transitive     _≃_
≃-isTrans (rfl x) (rfl y) = rfl (≡.trans x y)
≃-isTrans (gnl x) (gnl y) = gnl (λ i → ≃-isTrans (x i) (y i))

≃-isEquiv : IsEquivalence _≃_
≃-isEquiv = record { refl = ≃-isRefl ; sym = ≃-isSym ; trans = ≃-isTrans }
```

We now define, for a given signature S and context X, the algebraic structure **T** X, known as the *term algebra in* S *over* X. Terms are viewed as acting on other terms, so both the elements of the domain of **T** X and its basic operations are the terms themselves. That is, for each operation symbol f : $\mid S \mid$, we denote by f ̂ **T** X the operation on Term X that maps each tuple of terms, say, t : $\parallel S \parallel$ f → Term X, to the formal term f t. We codify these notions in Agda as follows.

```
TermSetoid : (X : Type χ) → Setoid _ _
TermSetoid X = record { Carrier = Term X ; _≈_ = _≃_ ; isEquivalence = ≃-isEquiv }

T : (X : Type χ) → Algebra (ov χ) (ov χ)
Algebra.Domain (T X) = TermSetoid X
Algebra.Interp (T X) ⟨$⟩ (f , ts) = node f ts
cong (Algebra.Interp (T X)) (≡.refl , ss≃ts) = gnl ss≃ts
```

Substitution, environments and interpretation of terms

In this section, we formalize the notions of *substitution*, *environment*, and *interpretation of terms* in an algebra. The approach to formalizing these concepts, and the Agda code presented in this subsection, is based on similar code developed by Andreas Abel to formalize Birkhoff's completeness theorem [1].

Recall that the domain of an algebra **A** is a setoid, which we denote by 𝔻[**A**], whose Carrier is the carrier of the algebra, 𝕌[**A**], and whose equivalence relation represents equality of elements in 𝕌[**A**].

The function Sub performs substitution from one context to another. Specifically, if X and Y are contexts, then Sub X Y assigns a term in X to each symbol in Y. The definition of Sub is a slight modification of the one given by Andreas Abel (*op. cit.*), as is the recursive definition of [σ] t, which denotes a substitution applied to a term.

```
Sub : Type χ → Type χ → Type _
Sub X Y = (y : Y) → Term X

[_]_ : {X Y : Type χ} → Sub X Y → Term Y → Term X
[ σ ] (ℊ x) = σ x
[ σ ] (node f ts) = node f (λ i → [ σ ] (ts i))
```

Fix a signature S, a context X, and an S-algebra $\mathbf{A}$. An *environment* for these data consists of the function type X → 𝕌[**A**] along with an equality on this type. The function Env manifests this notion by taking an S-algebra $\mathbf{A}$ and a context X and returning a setoid whose Carrier is the type X → 𝕌[**A**] and whose equivalence relation is pointwise equality of functions in X → 𝕌[**A**].

```
module Environment (𝐀 : Algebra α ℓ) where
  open Setoid 𝔻[ 𝐀 ] using ( _≈_ ; refl ; sym ; trans )
  Env : Type χ → Setoid _ _
  Env X = record { Carrier = X → 𝕌[ 𝐀 ]
                 ; _≈_ = λ ρ τ → (x : X) → ρ x ≈ τ x
                 ; isEquivalence = record { refl  = λ _     → refl
                                          ; sym   = λ h x   → sym (h x)
                                          ; trans = λ g h x → trans (g x)(h x) }}
```

Notice that this definition, as well as the next, are relative to a certain fixed algebra, so we put them inside a submodule called Environment. This allows us to load the submodule and associate its definitions with a number of different algebras simultaneously.

Next, the recursive function ⟦_⟧ denotes *interpretation* of a term in a given algebra, *evaluated* in a given environment.

```
⟦_⟧ : {X : Type χ}(t : Term X) → (Env X) ⟶ 𝔻[ 𝐀 ]
⟦ ℊ x ⟧           ⟨$⟩ ρ       = ρ x
⟦ node f args ⟧ ⟨$⟩ ρ       = (Interp 𝐀) ⟨$⟩ (f , λ i → ⟦ args i ⟧ ⟨$⟩ ρ)
cong ⟦ ℊ x ⟧ u≈v            = u≈v x
cong ⟦ node f args ⟧ x≈y = cong (Interp 𝐀)(≡.refl , λ i → cong ⟦ args i ⟧ x≈y )
```

Two terms interpreted in $\mathbf{A}$ are proclaimed *equal* if they are equal for all environments. This equivalence of terms, and proof that it is an equivalence relation, is formalized in Agda as follows.

```
Equal : {X : Type χ}(s t : Term X) → Type _
Equal {X = X} s t = ∀ (ρ : Carrier (Env X)) → ⟦ s ⟧ ⟨$⟩ ρ ≈ ⟦ t ⟧ ⟨$⟩ ρ

≃→Equal : {X : Type χ}(s t : Term X) → s ≃ t → Equal s t
≃→Equal .(ℊ _) .(ℊ _) (rfl ≡.refl) = λ _ → refl
≃→Equal (node _ s)(node _ t)(gnl x) =
  λ ρ → cong (Interp 𝐀)(≡.refl , λ i → ≃→Equal(s i)(t i)(x i)ρ )

EqualIsEquiv : {Γ : Type χ} → IsEquivalence (Equal {X = Γ})
reflᵉ  EqualIsEquiv = λ _     → refl
symᵉ   EqualIsEquiv = λ x=y ρ → sym (x=y ρ)
transᵉ EqualIsEquiv = λ ij jk ρ → trans (ij ρ) (jk ρ)
```

The next lemma says that applying a substitution σ to a term t and evaluating the result

in the environment ρ has the same effect as evaluating t the a new environment, specifically, in the environment λ x → ⟦ σ x ⟧ ⟨$⟩ ρ (see [1] or [12, Lem. 3.3.11]).

```
substitution : {X Y : Type χ} → (t : Term Y) (σ : Sub X Y) (ρ : Carrier( Env X ) )
  →            ⟦ [ σ ] t ⟧ ⟨$⟩ ρ ≈ ⟦ t ⟧ ⟨$⟩ (λ x → ⟦ σ x ⟧ ⟨$⟩ ρ)
substitution (ℊ x)       σ ρ = refl
substitution (node f ts) σ ρ = cong (Interp A)(≡.refl , λ i → substitution (ts i) σ ρ)
```

This concludes the definition of the Environment module based on [1].

Compatibility of terms

We will need two more facts about term operations. The first, called comm-hom-term, asserts that every term commutes with every homomorphism. The second, interp-prod, shows how to express the interpretation of a term in a product algebra.

```
module _ {X : Type χ}{A : Algebra α ρᵃ}{B : Algebra β ρᵇ}(hh : hom A B) where
  open Environment A using ( ⟦_⟧ )
  open Environment B using () renaming ( ⟦_⟧ to ⟦_⟧ᴮ )
  open Setoid 𝔻[ B ] using ( _≈_ ; refl )
  private hfunc = ∣ hh ∣ ; h = _⟨$⟩_ hfunc
  comm-hom-term : (t : Term X) (a : X → 𝕌[ A ]) → h (⟦ t ⟧ ⟨$⟩ a) ≈ ⟦ t ⟧ᴮ ⟨$⟩ (h ∘ a)
  comm-hom-term (ℊ x) a      = refl
  comm-hom-term (node f t) a =
    begin
      h(⟦ node f t ⟧ ⟨$⟩ a)            ≈⟨ compatible ∥ hh ∥ ⟩
      (f ̂ B)(λ i → h(⟦ t i ⟧ ⟨$⟩ a)) ≈⟨ cong(Interp B)(≡.refl , λ i → comm-hom-term (t i) a)⟩
      ⟦ node f t ⟧ᴮ ⟨$⟩ (h ∘ a)       ■ where open SetoidReasoning 𝔻[ B ]

module _ {X : Type χ}{ι : Level} {I : Type ι} (𝒜 : I → Algebra α ρᵃ) where
  open Setoid 𝔻[ ⨅ 𝒜 ] using ( _≈_ )
  open Environment     using ( ⟦_⟧ ; ≃→Equal )
  interp-prod : (p : Term X) → ∀ ρ → (⟦ ⨅ 𝒜 ⟧ p) ⟨$⟩ ρ ≈ λ i → (⟦ 𝒜 i ⟧ p) ⟨$⟩ λ x → (ρ x) i
  interp-prod (ℊ x)      = λ ρ i → ≃→Equal (𝒜 i) (ℊ x) (ℊ x) ≃-isRefl λ _ → (ρ x) i
  interp-prod (node f t) = λ ρ   → cong (Interp (⨅ 𝒜)) ( ≡.refl , λ j k → interp-prod (t j) ρ k )
```

4 Equational Logic

Term identities, equational theories, and the ⊨ relation

Given a signature S and a context X, an *S-term equation* or *S-term identity* is an ordered pair (p , q) of S-terms. For instance, if the context is X : Type χ, then a term equation is a pair inhabiting the Cartesian product type Term X × Term X. Such pairs of terms are also denoted by p ≈ q and are often simply called equations or identities, especially when the signature S is obvious.

We define an *equational theory* (or *algebraic theory*) to be a pair T = (S , $\mathscr{E}^T$) consisting of a signature S and a collection $\mathscr{E}^T$ of S-term equations. Some authors reserve the term *theory* for a *deductively closed* set of equations, that is, a set of equations that is closed under *entailment* (defined below).

We say that the algebra **A** *satisfies* the equation p ≈ q if, for all ρ : X → 𝔻[**A**], we have ⟦ p ⟧ ⟨$⟩ ρ ≈ ⟦ q ⟧ ⟨$⟩ ρ. In other words, when they are interpreted in the algebra **A**, the terms p and q are equal no matter what values in **A** are assigned to variable symbols in X.

In this situation, we write $\mathbf{A} \models \mathsf{p} \approx \mathsf{q}$ and say that $\mathbf{A}$ *models* $\mathsf{p} \approx \mathsf{q}$, or that $\mathbf{A}$ is a *model* of $\mathsf{p} \approx \mathsf{q}$. If $\mathscr{K}$ is a class of algebras, all of the same signature, we write $\mathscr{K} \Vdash \mathsf{p} \approx \mathsf{q}$ and say that $\mathscr{K}$ *models* the identity $\mathsf{p} \approx \mathsf{q}$ provided for every $\mathbf{A} \in \mathscr{K}$, we have $\mathbf{A} \models \mathsf{p} \approx \mathsf{q}$.

```
module _ {X : Type χ} where

_⊨_≈_ : Algebra α ρᵃ → Term X → Term X → Type _
𝐀 ⊨ p ≈ q = Equal p q where open Environment 𝐀

_||⊨_≈_ : Pred (Algebra α ρᵃ) ℓ → Term X → Term X → Type _
𝒦 ||⊨ p ≈ q = ∀ 𝐀 → 𝒦 𝐀 → 𝐀 ⊨ p ≈ q
```

We represent a set of identities as a predicate over pairs of terms, say, $\mathscr{E}$: `Pred (Term X × Term X) _` and we denote by $\mathbf{A} \models\!\models \mathscr{E}$ the assertion that the algebra $\mathbf{A}$ models $\mathsf{p} \approx \mathsf{q}$ for all $(\mathsf{p}\ ,\ \mathsf{q}) \in \mathscr{E}$.[10]

```
_⊨⊨_ : (𝐀 : Algebra α ρᵃ) → Pred(Term X × Term X)(ov χ) → Type _
𝐀 ⊨⊨ ℰ = ∀ {p q} → (p , q) ∈ ℰ → Equal p q where open Environment 𝐀
```

If $\mathscr{K}$ is a class of structures and $\mathscr{E}$ a set of term identities, then the set of term equations modeled by $\mathscr{K}$ is denoted by `Th` $\mathscr{K}$ and is called the *equational theory* of $\mathscr{K}$, while the class of structures modeling $\mathscr{E}$ is denoted by `Mod` $\mathscr{E}$ and is called the *equational class axiomatized* by $\mathscr{E}$. We formalize these concepts in Agda with the following types.

```
Th : {X : Type χ} → Pred (Algebra α ρᵃ) ℓ → Pred(Term X × Term X) _
Th 𝒦 = λ (p , q) → 𝒦 ||⊨ p ≈ q

Mod : {X : Type χ} → Pred(Term X × Term X) ℓ → Pred (Algebra α ρᵃ) _
Mod ℰ 𝐀 = ∀ {p q} → (p , q) ∈ ℰ → Equal p q where open Environment 𝐀
```

Entailment

If $\mathscr{E}$ is a set of S-term equations and p and q are S-terms, we say that $\mathscr{E}$ *entails* the equation $\mathsf{p} \approx \mathsf{q}$, and we write $\mathscr{E} \vdash \mathsf{p} \approx \mathsf{q}$, just in case every model of $\mathscr{E}$ also models $\mathsf{p} \approx \mathsf{q}$. We represent entailment in type theory using an inductive type that is similar to the one defined by Abel in [1]. We call this the *entailment type* and define it as follows.

```
data _⊢_▷_≈_ (ℰ : {Y : Type χ} → Pred(Term Y × Term Y) (ov χ)) :
             (X : Type χ)(p q : Term X) → Type (ov χ) where

  hyp       : ∀{Y}{p q : Term Y} → (p , q) ∈ ℰ → ℰ ⊢ _ ▷ p ≈ q
  app       : ∀{Y}{ps qs : ∥ 𝑆 ∥ 𝑓 → Term Y}
                        → (∀ i → ℰ ⊢ Y ▷ ps i ≈ qs i) → ℰ ⊢ Y ▷ (node 𝑓 ps) ≈ (node 𝑓 qs)
  sub       : ∀{p q}    → ℰ ⊢ Γ ▷ p ≈ q → (σ : Sub Δ Γ) → ℰ ⊢ Δ ▷ ([ σ ] p) ≈ ([ σ ] q)
  reflexive : ∀{p}      → ℰ ⊢ Γ ▷ p ≈ p
  symmetric : ∀{p q}    → ℰ ⊢ Γ ▷ p ≈ q → ℰ ⊢ Γ ▷ q ≈ p
  transitive : ∀{p q r} → ℰ ⊢ Γ ▷ p ≈ q → ℰ ⊢ Γ ▷ q ≈ r → ℰ ⊢ Γ ▷ p ≈ r
```

[10] Notice that $\models\!\models$ is a stretched version of the models symbol, $\models$; this makes it possible for Agda to distinguish and parse expressions involving the types `_⊨⊨_` and `_⊨_≈_`. In Emacs `agda2-mode`, the symbol $\models\!\models$ is produced by typing `\|=`, while $\models$ is produced with `\models`.

The fact that this type represents the informal semantic notion of entailment given at the start of this subsection is called *soundness* and *completeness*. More precisely, *the entailment type is sound* means the following: if $\mathscr{E} \vdash X \triangleright p \approx q$, then $p \approx q$ holds in every model of $\mathscr{E}$. *The entailment type is complete* means the following: if $p \approx q$ holds in every model of $\mathscr{E}$, then $\mathscr{E} \vdash X \triangleright p \approx q$. Soundness and completeness of an entailment type similar to the one defined above was proved by Abel in [1]. We will invoke soundness of the entailment type only once below; nonetheless, here is its formalization (due to Abel, *op. cit.*):

```
module Soundness (ℰ : {Y : Type χ} → Pred(Term Y × Term Y) (ov χ))
                 (𝐀 : Algebra α ρᵃ)                    -- We assume an algebra 𝐀
                 (V : ∀{Y} → _╠═══_{χ = χ} 𝐀 (ℰ{Y})) -- that models all equations in ℰ.
                 where
  open SetoidReasoning 𝔻[ 𝐀 ]
  open Environment 𝐀
  sound : ∀ {p q} → ℰ ⊢ Γ ▷ p ≈ q → 𝐀 ⊨ p ≈ q
  sound (hyp i) = V i
  sound (app es) ρ = cong (Interp 𝐀) (≡.refl , λ i → sound (es i) ρ)
  sound (sub {p = p}{q} Epq σ) ρ =
    begin
      ⟦ [ σ ] p ⟧ ⟨$⟩                    ρ   ≈⟨  substitution p σ ρ                ⟩
      ⟦ p       ⟧ ⟨$⟩ (λ x → ⟦ σ x ⟧ ⟨$⟩ ρ)  ≈⟨  sound Epq (λ x → ⟦ σ x ⟧ ⟨$⟩ ρ) ⟩
      ⟦ q       ⟧ ⟨$⟩ (λ x → ⟦ σ x ⟧ ⟨$⟩ ρ)  ≈˘⟨ substitution q σ ρ                ⟩
      ⟦ [ σ ] q ⟧ ⟨$⟩                    ρ   ■
  sound (reflexive   {p = p}                ) = reflᵉ   EqualIsEquiv {x = p}
  sound (symmetric {p = p}{q}     Epq       ) = symᵉ    EqualIsEquiv {x = p}{q}     (sound Epq)
  sound (transitive  {p = p}{q}{r} Epq Eqr ) = transᵉ  EqualIsEquiv {i = p}{q}{r} (sound Epq)(sound Eqr)
```

The Closure Operators H, S, P and V

Fix a signature S, let $\mathscr{K}$ be a class of S-algebras, and define

- H $\mathscr{K}$ = algebras isomorphic to homomorphic images of members of $\mathscr{K}$;
- S $\mathscr{K}$ = algebras isomorphic to subalgebras of a members of $\mathscr{K}$;
- P $\mathscr{K}$ = algebras isomorphic to products of members of $\mathscr{K}$.

A straight-forward verification confirms that H, S, and P are *closure operators* (expansive, monotone, and idempotent). A class $\mathscr{K}$ of S-algebras is said to be *closed under the taking of homomorphic images* provided H $\mathscr{K} \subseteq \mathscr{K}$. Similarly, $\mathscr{K}$ is *closed under the taking of subalgebras* (resp., *arbitrary products*) provided S $\mathscr{K} \subseteq \mathscr{K}$ (resp., P $\mathscr{K} \subseteq \mathscr{K}$). The operators H, S, and P can be composed with one another repeatedly, forming yet more closure operators.

A *variety* is a class of S-algebras that is closed under the taking of homomorphic images, subalgebras, and arbitrary products. To represent varieties we define types for the closure operators H, S, and P that are composable; we then define a type V which represents closure under all three of these operators. Thus, if $\mathscr{K}$ is a class of S-algebras, then V $\mathscr{K}$:= H (S (P $\mathscr{K}$)), and $\mathscr{K}$ is a variety iff V $\mathscr{K} \subseteq \mathscr{K}$.

We now define the type H to represent classes of algebras that include all homomorphic images of algebras in the class—i.e., classes that are closed under the taking of homomorphic images—the type S to represent classes of algebras that closed under the taking of subalgebras, and the type P to represent classes of algebras closed under the taking of arbitrary products.

```
  module _ {α ρᵃ β ρᵇ : Level} where

  private a = α ⊔ ρᵃ
```

```
H : ∀ ℓ → Pred(Algebra α ρᵃ) (a ⊔ ov ℓ) → Pred(Algebra β ρᵇ) _
H _ 𝒦 B = Σ[ A ∈ Algebra α ρᵃ ] A ∈ 𝒦 × B IsHomImageOf A

S : ∀ ℓ → Pred(Algebra α ρᵃ) (a ⊔ ov ℓ) → Pred(Algebra β ρᵇ) _
S _ 𝒦 B = Σ[ A ∈ Algebra α ρᵃ ] A ∈ 𝒦 × B ≤ A

P : ∀ ℓ ι → Pred(Algebra α ρᵃ) (a ⊔ ov ℓ) → Pred(Algebra β ρᵇ) _
P _ ι 𝒦 B = Σ[ I ∈ Type ι ] (Σ[ 𝒜 ∈ (I → Algebra α ρᵃ) ] (∀ i → 𝒜 i ∈ 𝒦) × (B ≅ ⨅ 𝒜))
```

Finally, we define the *varietal closure* of a class $\mathscr{K}$ to be the class $\mathbf{V}\,\mathscr{K} := \mathbf{H}\,(\mathbf{S}\,(\mathbf{P}\,\mathscr{K}))$.

```
module _ {α ρᵃ β ρᵇ γ ρᶜ δ ρᵈ : Level} where

private a = α ⊔ ρᵃ ; b = β ⊔ ρᵇ
V : ∀ ℓ ι → Pred(Algebra α ρᵃ) (a ⊔ ov ℓ) → Pred(Algebra δ ρᵈ) _
V ℓ ι 𝒦 = H{γ}{ρᶜ}{δ}{ρᵈ} (a ⊔ b ⊔ ℓ ⊔ ι) (S{β}{ρᵇ} (a ⊔ ℓ ⊔ ι) (P ℓ ι 𝒦))
```

An important property of the binary relation $\models$ is *algebraic invariance* (i.e., invariance under isomorphism). We formalize this property as follows.

```
module _ {X : Type χ}{A : Algebra α ρᵃ}(B : Algebra β ρᵇ)(p q : Term X) where

⊧-I-invar : A ⊧ p ≈ q → A ≅ B → B ⊧ p ≈ q

⊧-I-invar Apq (mkiso fh gh f∼g g∼f) ρ =
  begin
      ⟦ p ⟧  ⟨$⟩             ρ  ≈˘⟨ cong ⟦ p ⟧ (f∼g ∘ ρ)          ⟩
      ⟦ p ⟧  ⟨$⟩ (f ∘  (g ∘ ρ)) ≈˘⟨ comm-hom-term fh p (g ∘ ρ) ⟩
    f(⟦ p ⟧ᴬ ⟨$⟩       (g ∘ ρ)) ≈⟨  cong | fh | (Apq (g ∘ ρ))    ⟩
    f(⟦ q ⟧ᴬ ⟨$⟩       (g ∘ ρ)) ≈⟨  comm-hom-term fh q (g ∘ ρ) ⟩
      ⟦ q ⟧  ⟨$⟩ (f ∘  (g ∘ ρ)) ≈⟨  cong ⟦ q ⟧ (f∼g ∘ ρ)          ⟩
      ⟦ q ⟧  ⟨$⟩             ρ  ■
  where
  private f = _⟨$⟩_ | fh | ; g = _⟨$⟩_ | gh |
  open Environment A using () renaming ( ⟦_⟧ to ⟦_⟧ᴬ )
  open Environment B using ( ⟦_⟧ )
  open SetoidReasoning 𝔻[ B ]
```

Identities modeled by an algebra $\mathbf{A}$ are also modeled by every homomorphic image of $\mathbf{A}$ and by every subalgebra of $\mathbf{A}$. These facts are formalized in Agda as follows.

```
module _ {X : Type χ}{A : Algebra α ρᵃ}{B : Algebra β ρᵇ}{p q : Term X} where

  ⊧-H-invar : A ⊧ p ≈ q → B IsHomImageOf A → B ⊧ p ≈ q
  ⊧-H-invar Apq (φh , φE) ρ =
    begin
        ⟦ p ⟧  ⟨$⟩                ρ  ≈˘⟨ cong ⟦ p ⟧(λ _ → InvIsInverseʳ φE) ⟩
        ⟦ p ⟧  ⟨$⟩ (φ ∘ φ⁻¹ ∘ ρ)  ≈˘⟨ comm-hom-term φh p (φ⁻¹ ∘ ρ)     ⟩
      φ(⟦ p ⟧ᴬ ⟨$⟩ (    φ⁻¹ ∘ ρ)) ≈⟨  cong | φh | (Apq (φ⁻¹ ∘ ρ))      ⟩
      φ(⟦ q ⟧ᴬ ⟨$⟩ (    φ⁻¹ ∘ ρ)) ≈⟨  comm-hom-term φh q (φ⁻¹ ∘ ρ)     ⟩
        ⟦ q ⟧  ⟨$⟩ (φ ∘ φ⁻¹ ∘ ρ)  ≈⟨  cong ⟦ q ⟧(λ _ → InvIsInverseʳ φE) ⟩
        ⟦ q ⟧  ⟨$⟩                ρ  ■
    where
    φ⁻¹ : 𝕌[ B ] → 𝕌[ A ]
    φ⁻¹ = SurjInv | φh | φE
```

```
    private φ = (_⟨$⟩_ | φh |)
    open Environment 𝐀 using () renaming ( ⟦_⟧ to ⟦_⟧^A)
    open Environment 𝐁 using ( ⟦_⟧ )
    open SetoidReasoning 𝔻[ 𝐁 ]

  ⊧-S-invar : 𝐀 ⊧ p ≈ q → 𝐁 ≤ 𝐀 → 𝐁 ⊧ p ≈ q
  ⊧-S-invar Apq B≤A b = ∥ B≤A ∥
    ( begin
      h ( ⟦ p ⟧   ⟨$⟩       b) ≈⟨  comm-hom-term hh p b ⟩
          ⟦ p ⟧^A ⟨$⟩ (h ∘ b) ≈⟨  Apq (h ∘ b)            ⟩
          ⟦ q ⟧^A ⟨$⟩ (h ∘ b) ≈˘⟨ comm-hom-term hh q b ⟩
      h ( ⟦ q ⟧   ⟨$⟩       b) ■ )
    where
    open SetoidReasoning 𝔻[ 𝐀 ]
    open Setoid 𝔻[ 𝐀 ]   using ( _≈_ )
    open Environment 𝐀 using () renaming ( ⟦_⟧ to ⟦_⟧^A )
    open Environment 𝐁 using ( ⟦_⟧ )
    private hh = | B≤A | ; h = _⟨$⟩_ | hh |
```

An identity satisfied by all algebras in an indexed collection is also satisfied by the product of algebras in the collection.

```
module _ {X : Type χ}{I : Type ℓ}(𝒜 : I → Algebra α ρ^a){p q : Term X} where
  ⊧-P-invar : (∀ i → 𝒜 i ⊧ p ≈ q) → ⨅ 𝒜 ⊧ p ≈ q
  ⊧-P-invar 𝒜pq a =
    begin
      ⟦ p ⟧₁                ⟨$⟩ a                ≈⟨  interp-prod 𝒜 p a ⟩
      ( λ i → (⟦ 𝒜 i ⟧ p) ⟨$⟩ λ x → (a x) i ) ≈⟨ (λ i → 𝒜pq i (λ x → (a x) i)) ⟩
      ( λ i → (⟦ 𝒜 i ⟧ q) ⟨$⟩ λ x → (a x) i ) ≈˘⟨ interp-prod 𝒜 q a ⟩
      ⟦ q ⟧₁                ⟨$⟩ a                ■
    where
    open Environment (⨅ 𝒜) using () renaming ( ⟦_⟧ to ⟦_⟧₁ )
    open Environment          using ( ⟦_⟧ )
    open Setoid 𝔻[ ⨅ 𝒜 ]      using ( _≈_ )
    open SetoidReasoning 𝔻[ ⨅ 𝒜 ]
```

The classes H $\mathscr{K}$, S $\mathscr{K}$, P $\mathscr{K}$, and V $\mathscr{K}$ all satisfy the same term identities. We will only use a subset of the inclusions needed to prove this assertion, and we present here only the facts we need.[11] First, the closure operator H preserves the identities modeled by the given class; this follows almost immediately from the invariance lemma ⊧-H-invar proved above.

```
module _ {X : Type χ}{𝒦 : Pred(Algebra α ρ^a) (α ⊔ ρ^a ⊔ ov ℓ)}{p q : Term X} where
  H-id1 : 𝒦 ∣⊧ p ≈ q → H{β = α}{ρ^a}ℓ 𝒦 ∣⊧ p ≈ q
  H-id1 σ 𝐁 (𝐀 , kA , BimgA) = ⊧-H-invar{p = p}{q} (σ 𝐀 kA) BimgA
```

The analogous preservation result for S is a simple consequence of the invariance lemma ⊧-S-invar; the obvious converse, which we call S-id2, has an equally straightforward proof.

```
  S-id1 : 𝒦 ∣⊧ p ≈ q → S{β = α}{ρ^a}ℓ 𝒦 ∣⊧ p ≈ q
  S-id1 σ 𝐁 (𝐀 , kA , B≤A) = ⊧-S-invar{p = p}{q} (σ 𝐀 kA) B≤A
  S-id2 : S ℓ 𝒦 ∣⊧ p ≈ q → 𝒦 ∣⊧ p ≈ q
```

[11] For more details, see the Varieties.Func.Preservation module of the agda-algebras library.

```
  S-id2 Spq 𝐀 kA = Spq 𝐀 (𝐀 , (kA , ≤-reflexive))
```

Finally, we have analogous pairs of implications for P and V, In each case, we will only need the first implication, so we omit the others from this presentation.

```
  P-id1 : ∀{ι} → 𝒦 ||≈ p ≈ q → P{β = α}{ρᵃ}ℓ ι 𝒦 ||≈ p ≈ q
  P-id1 σ 𝐀 (I , 𝒜 , kA , A≅⨅A) = ⊧-I-invar 𝐀 p q IH (≅-sym A≅⨅A)
    where
    IH : ⨅ 𝒜 ⊧ p ≈ q
    IH = ⊧-P-invar 𝒜 {p}{q} (λ i → σ (𝒜 i) (kA i))

module _ {X : Type χ}{ι : Level}{𝒦 : Pred(Algebra α ρᵃ)(α ⊔ ρᵃ ⊔ ov ℓ)}{p q : Term X} where
  private aℓι = α ⊔ ρᵃ ⊔ ℓ ⊔ ι

  V-id1 : 𝒦 ||≈ p ≈ q → V ℓ ι 𝒦 ||≈ p ≈ q
  V-id1 σ 𝐁 (𝐀 , (⨅A , p⨅A , A≤⨅A) , BimgA) =
    H-id1{ℓ = aℓι}{𝒦 = S aℓι (P {β = α}{ρᵃ}ℓ ι 𝒦)}{p = p}{q} spK⊧pq 𝐁 (𝐀 , (spA , BimgA))
      where
      spA : 𝐀 ∈ S aℓι (P {β = α}{ρᵃ}ℓ ι 𝒦)
      spA = ⨅A , (p⨅A , A≤⨅A)
      spK⊧pq : S aℓι (P ℓ ι 𝒦) ||≈ p ≈ q
      spK⊧pq = S-id1{ℓ = aℓι}{p = p}{q} (P-id1{ℓ = ℓ} {𝒦 = 𝒦}{p = p}{q} σ)
```

5 Free Algebras

The absolutely free algebra

The term algebra $\mathbf{T}\ X$ is *absolutely free* (or *universal*, or *initial*) for algebras in the signature S. That is, for every S-algebra $\mathbf{A}$, the following hold.

- Every function from X to 𝕌[𝐀] lifts to a homomorphism from $\mathbf{T}\ X$ to $\mathbf{A}$.
- The homomorphism that exists by the previous item is unique.

We now prove the first of these facts in Agda which we call free-lift.[12,13]

```
  module _ {X : Type χ}{𝐀 : Algebra α ρᵃ}(h : X → 𝕌[ 𝐀 ]) where
    free-lift : 𝕌[ 𝐓 X ] → 𝕌[ 𝐀 ]
    free-lift (ℊ x) = h x
    free-lift (node f t) = (f ̂ 𝐀) (λ i → free-lift (t i))

  free-lift-func : 𝔻[ 𝐓 X ] ⟶ 𝔻[ 𝐀 ]
  free-lift-func ⟨$⟩ x = free-lift x
  cong free-lift-func = fl-cong
    where
    open Setoid 𝔻[ 𝐀 ] using ( _≈_ ) renaming ( reflexive to reflexiveᴬ )
    fl-cong : ∀ {s t} → s ≃ t → free-lift s ≈ free-lift t
    fl-cong (_≃_.rfl x) = reflexiveᴬ (≡.cong h x)
    fl-cong (_≃_.gnl x) = cong (Interp 𝐀) (≡.refl , (λ i → fl-cong (x i)))
```

[12] The agda-algebras library also defines free-lift-func : 𝔻[𝐓 X] ⟶ 𝔻[𝐀] for constructing the analogous setoid function.

[13] For the proof of uniqueness, see the Terms.Func.Properties module of the agda-algebras library.

Evidently, the proof is a straightforward structural induction argument. At the base step, when the term has the form $\mathcal{g}$ x, the free lift of h agrees with h; at the inductive step, when the term has the form node f t, we assume (the induction hypothesis) that the image of each subterm t i under the free lift of h is known and the free lift is defined by applying f ̂ **A** to these images. Moreover, the free lift so defined is a homomorphism by construction; indeed, here is the trivial proof.

```
lift-hom : hom (T X) A
lift-hom = free-lift-func , hhom
  where
  hfunc : 𝔻[ T X ] ⟶ 𝔻[ A ]
  hfunc = free-lift-func

  hcomp : compatible-map (T X) A free-lift-func
  hcomp {f}{a} = cong (Interp A) (≡.refl , (λ i → (cong free-lift-func){a i} ≃-isRefl))

  hhom : IsHom (T X) A hfunc
  hhom = mkhom (λ{f}{a} → hcomp{f}{a})
```

It turns out that the interpretation of a term p in an environment η is the same as the free lift of η evaluated at p.

```
module _ {X : Type χ}{A : Algebra α ρᵃ} where
  open Setoid 𝔻[ A ]   using ( _≈_ ; refl )
  open Environment A using ( ⟦_⟧ )

free-lift-interp : (η : X → 𝕌[ A ])(p : Term X) → ⟦ p ⟧ ⟨$⟩ η ≈ (free-lift{A = A} η) p
free-lift-interp η (ℊ x)      = refl
free-lift-interp η (node f t) = cong (Interp A) (≡.refl , (free-lift-interp η) ∘ t)
```

The relatively free algebra in theory

In this subsection, we describe, for a given class $\mathscr{K}$ of S-algebras, the *relatively free algebra* in S (P $\mathscr{K}$) over X, using the informal language that is typical of mathematics literature. In the next section we will present the relatively free algebra in Agda using the formal language of type theory.

Above we defined the term algebra **T** X, which is free in the class of all S-algebras; that is, **T** X has the universal property and belongs to the class of S-algebras. Given an arbitrary class $\mathscr{K}$ of S-algebras, we can't expect that **T** X belongs to $\mathscr{K}$, so, in general, we say that **T** X is free *for* $\mathscr{K}$. Indeed, it might not be possible to find a free algebra that belongs to $\mathscr{K}$. However, for any class $\mathscr{K}$ we can construct an algebra that is free for $\mathscr{K}$ and belongs to the class S (P $\mathscr{K}$), and for most applications this suffices.

The informal construction of the free algebra in S (P $\mathscr{K}$), for an arbitrary class $\mathscr{K}$ of S-algebras, proceeds by taking the quotient of **T** X modulo a congruence relation that we will denote by $\approx$. One approach is to let $\approx := \bigcap \{\theta \in \mathrm{Con}\ (\mathbf{T}\ X) : \mathbf{T}\ X\ /\ \theta \in S\ \mathscr{K}\ \}$.[14] Alternatively we could let $\mathscr{C} = \mathrm{Th}\ \mathscr{K}$ and take $\approx$ to be the least equivalence relation on the domain of **T** X such that

1. for every equation (p , q) $\in$ Th $\mathscr{K}$ and every environment ρ : X → Term X, we have ⟦ p ⟧ ⟨$⟩ $\rho \approx$ ⟦ q ⟧ ⟨$⟩ ρ, and

[14] Con (**T** X) is the set of congruences of **T** X.

2. $\approx$ is a congruence of $\mathbf{T}\ \mathsf{X}$; that is, for every operation symbol $\mathsf{f} : |\ S\ |$, and for all tuples $\mathsf{s}\ \mathsf{t} : \|\ S\ \| \mathsf{f} \to \mathsf{Term}\ \mathsf{X}$, the following implication holds:[15]

$$(\forall\ \mathsf{i} \to [\![\ \mathsf{s}\ \mathsf{i}\ ]\!]\ \langle\$\rangle\ \rho \approx [\![\ \mathsf{t}\ \mathsf{i}\ ]\!]\ \langle\$\rangle\ \rho) \to [\![\ \mathsf{f}\ \mathsf{s}\ ]\!]\ \langle\$\rangle\ \rho \approx [\![\ \mathsf{f}\ \mathsf{t}\ ]\!]\ \langle\$\rangle\ \rho$$

Whichever approach we choose, the *relatively free algebra over* X (relative to $\mathscr{K}$) is defined to be the quotient $\mathbb{F}[\ \mathsf{X}\ ] := \mathbf{T}\ \mathsf{X}\ /\ \approx$.

Evidently $\mathbb{F}[\ \mathsf{X}\ ]$ is a subdirect product of the algebras in $\{\mathbf{T}\ \mathsf{X}\ /\ \theta\}$, where θ ranges over congruences modulo which $\mathbf{T}\ \mathsf{X}$ belongs to $\mathsf{S}\ \mathscr{K}$. Thus, $\mathbb{F}[\ \mathsf{X}\ ] \in \mathsf{P}\ (\mathsf{S}\ \mathscr{K}) \subseteq \mathsf{S}\ (\mathsf{P}\ \mathscr{K})$, and it follows that $\mathbb{F}[\ \mathsf{X}\ ]$ satisfies the identities in $\mathsf{Th}\ \mathscr{K}$ (those modeled by all members of $\mathscr{K}$). Indeed, for each pair $\mathsf{p}\ \mathsf{q} : \mathsf{Term}\ \mathsf{X}$, if $\mathscr{K} \models \mathsf{p} \approx \mathsf{q}$, then p and q must belong to the same $\approx$-class, so p and q are identified in $\mathbb{F}[\ \mathsf{X}\ ]$. (Notice that $\approx$ may be empty, in which case $\mathbf{T}\ \mathsf{X}\ /\ \approx$ is trivial.)

The relatively free algebra in Agda

We now define the relatively free algebra in Agda using the language of type theory. Our approach will be different from the informal one described above, but the end result will be the same. We start with a type $\mathscr{E}$ representing a collection of identities and, instead of forming a quotient, we take the domain of the free algebra to be a setoid whose Carrier is the type $\mathsf{Term}\ \mathsf{X}$ of S-terms in X and whose equivalence relation includes all pairs $(\mathsf{p}\ ,\ \mathsf{q}) \in \mathsf{Term}\ \mathsf{X} \times \mathsf{Term}\ \mathsf{X}$ such that $\mathsf{p} \approx \mathsf{q}$ is derivable from $\mathscr{E}$; that is, $\mathscr{E} \vdash \mathsf{X} \vartriangleright \mathsf{p} \approx \mathsf{q}$. Observe that elements of this setoid are equal iff they belong to the same equivalence class of the congruence $\approx$ defined above. Therefore, the setoid so defined represents the quotient $\mathbf{T}\ \mathsf{X}\ /\ \approx$. Finally, the interpretation of an operation in the free algebra is simply the operation itself, which works since $\mathscr{E} \vdash \mathsf{X} \vartriangleright \_\approx\_$ is a congruence relation.

```
module FreeAlgebra {χ : Level}(ℰ : {Y : Type χ} → Pred (Term Y × Term Y) (ov χ)) where

  FreeDomain : Type χ → Setoid _ _
  FreeDomain X =
    record { Carrier       = Term X
           ; _≈_           = ℰ ⊢ X ▷_≈_
           ; isEquivalence = record { refl = reflexive ; sym = symmetric ; trans = transitive } }

  𝔽[_] : Type χ → Algebra (ov χ) _
  Domain 𝔽[ X ] = FreeDomain X
  Interp 𝔽[ X ] = FreeInterp where
    FreeInterp : ∀ {X} → ⟨ S ⟩ (FreeDomain X) ⟶ FreeDomain X
    FreeInterp ⟨$⟩ (f , ts)       = node f ts
    cong FreeInterp (≡.refl , h) = app h
```

The natural epimorphism

We now define the natural epimorphism from $\mathbf{T}\ \mathsf{X}$ onto the relatively free algebra $\mathbb{F}[\ \mathsf{X}\ ]$ and prove that its kernel is the congruence of $\mathbf{T}\ \mathsf{X}$ defined by the identities modeled by ($\mathsf{S}\ \mathscr{K}$, hence by) $\mathscr{K}$.

```
module FreeHom {𝒦 : Pred(Algebra α ρᵃ) (α ⊔ ρᵃ ⊔ ov ℓ)} where
  private c = α ⊔ ρᵃ ⊔ ℓ ; ι = ov c
  open FreeAlgebra {χ = c} (Th 𝒦) using ( 𝔽[_] )
```

[15] Here all interpretations, denoted by $[\![\_]\!]$, are with respect to $\mathbf{T}\ \mathsf{X}$.

```
  epiF[_] : (X : Type c) → epi (𝐓 X) 𝔽[ X ]
  epiF[ X ] = h , hepi
    where
    open Setoid 𝔻[ 𝐓 X ]   using ()       renaming ( _≈_ to _≈₀_ ; refl to refl^T )
    open Setoid 𝔻[ 𝔽[ X ] ] using ( refl ) renaming ( _≈_ to _≈₁_ )

    con : ∀ {x y} → x ≈₀ y → x ≈₁ y
    con (rfl {x}{y} ≡.refl) = refl
    con (gnl {f}{s}{t} x) = cong (Interp 𝔽[ X ]) (≡.refl , con ∘ x)

    h : 𝔻[ 𝐓 X ] ⟶ 𝔻[ 𝔽[ X ] ]
    h = record { f = id ; cong = con }

    hepi : IsEpi (𝐓 X) 𝔽[ X ] h
    compatible (isHom hepi) = cong h refl^T
    isSurjective hepi {y} = eq y refl

  homF[_] : (X : Type c) → hom (𝐓 X) 𝔽[ X ]
  homF[ X ] = IsEpi.HomReduct ∥ epiF[ X ] ∥

  kernel-in-theory : {X : Type c} → ker ∣ homF[ X ] ∣ ⊆ Th (V ℓ ι 𝒦)
  kernel-in-theory {X = X} {p , q} pKq 𝐀 vkA = V-id1{ℓ = ℓ}{p = p}{q} (ζ pKq) 𝐀 vkA
    where
    ζ : ∀{p q} → (Th 𝒦) ⊢ X ▷ p ≈ q → 𝒦 ∣⊨ p ≈ q
    ζ x 𝐀 kA = sound (λ y ρ → y 𝐀 kA ρ) x where open Soundness (Th 𝒦) 𝐀
```

Next we prove an important property of the relatively free algebra (relative to $\mathscr{K}$ and satisfying the identities in Th $\mathscr{K}$), which will be used in the formalization of the HSP theorem; this is the assertion that for every algebra $\mathbf{A}$, if $\mathbf{A} \models$ Th (V $\mathscr{K}$), then there exists an epimorphism from $\mathbb{F}[$ A $]$ onto $\mathbf{A}$.

```
module _ {𝐀 : Algebra (α ⊔ ρ^a ⊔ ℓ) (α ⊔ ρ^a ⊔ ℓ)} {𝒦 : Pred(Algebra α ρ^a) (α ⊔ ρ^a ⊔ ov ℓ)} where
  private c = α ⊔ ρ^a ⊔ ℓ ; ι = ov c
  open FreeHom {ℓ = ℓ} {𝒦}
  open FreeAlgebra {χ = c}(Th 𝒦) using ( 𝔽[_] )
  open Setoid 𝔻[ 𝐀 ]              using ( refl ; sym ; trans ) renaming ( Carrier to A )

    F-ModTh-epi : 𝐀 ∈ Mod (Th (V ℓ ι 𝒦)) → epi 𝔽[ A ] 𝐀
    F-ModTh-epi A∈ModThK = φ , isEpi
      where
      φ : 𝔻[ 𝔽[ A ] ] ⟶ 𝔻[ 𝐀 ]
      _⟨$⟩_ φ = free-lift{𝐀 = 𝐀} id
      cong φ {p} {q} pq = trans ( sym (free-lift-interp{𝐀 = 𝐀} id p) )
                                ( trans ( A∈ModThK{p = p}{q} (kernel-in-theory pq) id )
                                        ( free-lift-interp{𝐀 = 𝐀} id q ) )
      isEpi : IsEpi 𝔽[ A ] 𝐀 φ
      compatible (isHom isEpi) = cong (Interp 𝐀) (≡.refl , (λ _ → refl))
      isSurjective isEpi {y} = eq (g y) refl
```

Actually, we will need the following lifted version of this result.

```
F-ModTh-epi-lift : 𝐀 ∈ Mod (Th (V ℓ ι 𝒦)) → epi 𝔽[ A ] (Lift-Alg 𝐀 ι ι)
F-ModTh-epi-lift A∈ModThK = ∘-epi (F-ModTh-epi (λ {p q} → A∈ModThK{p = p}{q})) ToLift-epi
```

6 Birkhoff's Variety Theorem

Birkhoff's variety theorem, also known as the HSP theorem, asserts that a class of algebras is a variety if and only if it is an equational class. In this section, we present the statement and proof of the HSP theorem—first in the familiar, informal style similar to what one finds in standard textbooks (see, e.g., [3, Theorem 4.41]), and then in the formal language of Martin-Löf type theory using Agda.

6.1 Informal proof

Let $\mathscr{K}$ be a class of algebras and recall that $\mathscr{K}$ is a *variety* provided it is closed under homomorphisms, subalgebras and products; equivalently, $\mathsf{V}\ \mathscr{K} \subseteq \mathscr{K}$.[16] We call $\mathscr{K}$ an *equational class* if it is precisely the class of all models of some set of identities.

It is easy to prove that *every equational class is a variety*. Indeed, suppose $\mathscr{K}$ is an equational class axiomatized by the set $\mathscr{E}$ of term identities; that is, $\mathbf{A} \in \mathscr{K}$ iff $\mathbf{A} \models \mathscr{E}$. Since the classes $\mathsf{H}\ \mathscr{K}$, $\mathsf{S}\ \mathscr{K}$, $\mathsf{P}\ \mathscr{K}$ and $\mathscr{K}$ all satisfy the same set of equations, we have $\mathsf{V}\ \mathscr{K} \models \mathsf{p} \approx \mathsf{q}$ for all $(\mathsf{p}\ ,\ \mathsf{q}) \in \mathscr{E}$, so $\mathsf{V}\ \mathscr{K} \subseteq \mathscr{K}$.

The converse assertion—that *every variety is an equational class*—is less obvious. Let $\mathscr{K}$ be an arbitrary variety. We will describe a set of equations that axiomatizes $\mathscr{K}$. A natural choice is the set $\mathsf{Th}\ \mathscr{K}$ of all equations that hold in $\mathscr{K}$. Define $\mathscr{K}^+ = \mathsf{Mod}\ (\mathsf{Th}\ \mathscr{K})$. Clearly, $\mathscr{K} \subseteq \mathscr{K}^+$. We prove the reverse inclusion. Let $\mathbf{A} \in \mathscr{K}^+$; it suffices to find an algebra $\mathbf{F} \in \mathsf{S}\ (\mathsf{P}\ \mathscr{K})$ such that $\mathbf{A}$ is a homomorphic image of $\mathbf{F}$, as this will show that $\mathbf{A} \in \mathsf{H}\ (\mathsf{S}\ (\mathsf{P}\ \mathscr{K})) = \mathscr{K}$.

Let X be such that there exists a *surjective* environment $\rho : \mathsf{X} \to \mathbb{U}[\ \mathbf{A}\ ]$. By the lift-hom lemma, there is an epimorphism h from $\mathbf{T}\ \mathsf{X}$ onto $\mathbb{U}[\ \mathbf{A}\ ]$ that extends ρ. Now, put $\mathbb{F}[\ \mathsf{X}\ ] := \mathbf{T}\ \mathsf{X}/\Theta$, and let $\mathsf{g} : \mathbf{T}\ \mathsf{X} \to \mathbb{F}[\ \mathsf{X}\ ]$ be the natural epimorphism with kernel Θ. We claim that $\ker\ \mathsf{g} \subseteq \ker\ \mathsf{h}$. If the claim is true, then there is a map $\mathsf{f} : \mathbb{F}[\ \mathsf{X}\ ] \to \mathbf{A}$ such that $\mathsf{f} \circ \mathsf{g} = \mathsf{h}$. Since h is surjective, so is f. Hence $\mathbf{A} \in \mathsf{H}\ (\mathbb{F}\ \mathsf{X}) \subseteq \mathscr{K}^+$ completing the proof. To prove the claim, let $\mathsf{u}\ ,\ \mathsf{v} \in \mathbf{T}\ \mathsf{X}$ and assume that $\mathsf{g}\ \mathsf{u} = \mathsf{g}\ \mathsf{v}$. Since $\mathbf{T}\ \mathsf{X}$ is generated by X, there are terms $\mathsf{p}, \mathsf{q} \in \mathbf{T}\ \mathsf{X}$ such that $\mathsf{u} = ⟦\ \mathbf{T}\ \mathsf{X}\ ⟧\ \mathsf{p}$ and $\mathsf{v} = ⟦\ \mathbf{T}\ \mathsf{X}\ ⟧\ \mathsf{q}$.[17] Therefore,

$$⟦\ \mathbb{F}[\ \mathsf{X}\ ]\ ⟧\ \mathsf{p} = \mathsf{g}\ (⟦\ \mathbf{T}\ \mathsf{X}\ ⟧\ \mathsf{p}) = \mathsf{g}\ \mathsf{u} = \mathsf{g}\ \mathsf{v} = \mathsf{g}\ (⟦\ \mathbf{T}\ \mathsf{X}\ ⟧\ \mathsf{q}) = ⟦\ \mathbb{F}[\ \mathsf{X}\ ]\ ⟧\ \mathsf{q},$$

so $\mathscr{K} \models \mathsf{p} \approx \mathsf{q}$, so $(\mathsf{p}\ ,\ \mathsf{q}) \in \mathsf{Th}\ \mathscr{K}$. Since $\mathbf{A} \in \mathscr{K}^+ = \mathsf{Mod}\ (\mathsf{Th}\ \mathscr{K})$, we obtain $\mathbf{A} \models \mathsf{p} \approx \mathsf{q}$, so $\mathsf{h}\ \mathsf{u} = (⟦\ \mathbf{A}\ ⟧\ \mathsf{p})\ ⟨\$⟩\ \rho = (⟦\ \mathbf{A}\ ⟧\ \mathsf{q})\ ⟨\$⟩\ \rho = \mathsf{h}\ \mathsf{v}$, as desired.

6.2 Formal proof

We now show how to formally express and prove the twin assertions that (i) every equational class is a variety and (ii) every variety is an equational class.

Every equational class is a variety

For (i), we need an arbitrary equational class. To obtain one, we start with an arbitrary collection $\mathscr{E}$ of equations and let $\mathscr{K} = \mathsf{Mod}\ \mathscr{E}$, the equational class determined by $\mathscr{E}$. We prove that $\mathscr{K}$ is a variety by showing that $\mathscr{K} = \mathsf{V}\ \mathscr{K}$. The inclusion $\mathscr{K} \subseteq \mathsf{V}\ \mathscr{K}$, which holds for all classes $\mathscr{K}$, is called the *expansive* property of V. The converse inclusion $\mathsf{V}\ \mathscr{K} \subseteq \mathscr{K}$, on

[16] Recall, $\mathsf{V}\ \mathscr{K} := \mathsf{H}\ (\mathsf{S}\ (\mathsf{P}\ \mathscr{K}))$, and observe that $\mathscr{K} \subseteq \mathsf{V}\ \mathscr{K}$ holds for all $\mathscr{K}$ since V is a closure operator.
[17] Recall, $⟦\ \mathbf{A}\ ⟧\ \mathsf{t}$ denotes the interpretation of the term t in the algebra $\mathbf{A}$.

the other hand, requires the hypothesis that $\mathscr{K}$ is an equation class. We now formalize each of these inclusions.

```
module _ (𝒦 : Pred(Algebra α ρᵃ) (α ⊔ ρᵃ ⊔ ov ℓ))(X : Type (α ⊔ ρᵃ ⊔ ℓ)) where
  private ι = ov (α ⊔ ρᵃ ⊔ ℓ)

V-expa : 𝒦 ⊆ V ℓ ι 𝒦
V-expa {x = 𝐀} kA = 𝐀 ,(𝐀 ,(⊤ ,(λ _ → 𝐀), (λ _ → kA), Goal), ≤-reflexive), IdHomImage
  where
  open Setoid 𝔻[ 𝐀 ] using ( refl )
  open Setoid 𝔻[ ⨅ (λ _ → 𝐀) ] using () renaming ( refl to refl⨅ )

  to⨅ : 𝔻[ 𝐀 ] ⟶ 𝔻[ ⨅ (λ _ → 𝐀) ]
  (to⨅ ⟨$⟩ x) = λ _ → x
  cong to⨅ xy = λ _ → xy

  from⨅ : 𝔻[ ⨅ (λ _ → 𝐀) ] ⟶ 𝔻[ 𝐀 ]
  (from⨅ ⟨$⟩ x) = x tt
  cong from⨅ xy = xy tt

  Goal : 𝐀 ≅ ⨅ (λ x → 𝐀)
  Goal = mkiso (to⨅ , mkhom refl⨅) (from⨅ , mkhom refl) (λ _ _ → refl) (λ _ → refl)
```

Earlier we proved the identity preservation lemma, V-id1 : $\mathscr{K} \Vdash p \approx q \to V\ \ell\ \iota\ \mathscr{K} \Vdash p \approx q$. Thus, if $\mathscr{K}$ is an equational class, then $V\ \mathscr{K} \subseteq \mathscr{K}$, as we now confirm.

```
module _ {ℓ : Level}{X : Type ℓ}{ℰ : {Y : Type ℓ} → Pred (Term Y × Term Y) (ov ℓ)} where
  private 𝒦 = Mod{α = ℓ}{ℓ}{X} ℰ  -- an arbitrary equational class
  EqCl⇒Var : V ℓ (ov ℓ) 𝒦 ⊆ 𝒦
  EqCl⇒Var {𝐀}vA{p}{q} p℮q ρ = V-id1{ℓ = ℓ}{𝒦 = 𝒦}{p}{q}(λ _ x τ → x p℮q τ)𝐀 vA ρ
```

Together, V-expa and EqCl⇒Var prove that every equational class is a variety.

Every variety is an equational class

To prove statement (ii), we need an arbitrary variety; to obtain one, we start with an arbitrary class $\mathscr{K}$ of S-algebras and take its *varietal closure*, $V\ \mathscr{K}$. We prove that $V\ \mathscr{K}$ is an equational class by showing it is precisely the collection of algebras that model the equations in Th (V $\mathscr{K}$); that is, we prove $V\ \mathscr{K} = \text{Mod}\ (\text{Th}\ (V\ \mathscr{K}))$. The inclusion $V\ \mathscr{K} \subseteq \text{Mod}\ (\text{Th}\ (V\ \mathscr{K}))$ is a simple consequence of the fact that Mod Th is a closure operator. Nonetheless, completeness demands that we formalize this fact, however trivial is its proof.

```
module _ (𝒦 : Pred(Algebra α ρᵃ) (α ⊔ ρᵃ ⊔ ov ℓ))(X : Type (α ⊔ ρᵃ ⊔ ℓ)) where
  private c = α ⊔ ρᵃ ⊔ ℓ ; ι = ov c

ModTh-closure : V{β = β}{ρᵇ}{γ}{ρᶜ}{δ}{ρᵈ} ℓ ι 𝒦 ⊆ Mod{X = X} (Th (V ℓ ι 𝒦))
ModTh-closure {x = 𝐀} vA {p}{q} x ρ = x 𝐀 vA ρ
```

It remains to prove the converse inclusion, $\text{Mod}\ (\text{Th}\ (V\ \mathscr{K})) \subseteq V\ \mathscr{K}$, which is the main focus of the rest of the paper. We proceed as follows:

1. Let $\mathbf{C}$ be the product of all algebras in $S\ \mathscr{K}$, so that $\mathbf{C} \in P\ (S\ \mathscr{K})$.
2. Prove $P\ (S\ \mathscr{K}) \subseteq S\ (P\ \mathscr{K})$, so $\mathbf{C} \in S\ (P\ \mathscr{K})$ by item 1.
3. Prove $\mathbb{F}[\ X\ ] \leq \mathbf{C}$, so that $\mathbb{F}[\ X\ ] \in S\ (S\ (P\ \mathscr{K}))\ (= S\ (P\ \mathscr{K}))$.

4. Prove that every algebra in Mod (Th (V $\mathcal{K}$)) is a homomorphic image of $\mathbb{F}[\,\mathsf{X}\,]$ and thus belongs to H (S (P $\mathcal{K}$)) (= V $\mathcal{K}$).

To define $\mathbf{C}$ as the product of all algebras in S $\mathcal{K}$, we must first contrive an index type for the class S $\mathcal{K}$. We do so by letting the indices be the algebras belonging to $\mathcal{K}$. Actually, each index will consist of a triple ($\mathbf{A}$, p , ρ) where $\mathbf{A}$ is an algebra, p : $\mathbf{A} \in$ S $\mathcal{K}$) is a proof of membership in $\mathcal{K}$, ρ : X $\to$ $\mathbb{U}[\,\mathbf{A}\,]$ is an arbitrary environment. Using this indexing scheme, we construct $\mathbf{C}$, the product of all algebras in $\mathcal{K}$ and all environments, as follows.

```
open FreeHom {ℓ = ℓ} {𝒦}
open FreeAlgebra {χ = c}(Th 𝒦) using ( 𝔽[_] )
open Environment               using ( Env )

ℑ⁺ : Type ι
ℑ⁺ = Σ[ 𝐀 ∈ (Algebra α ρᵃ) ] (𝐀 ∈ S ℓ 𝒦) × (Carrier (Env 𝐀 X))

𝔄⁺ : ℑ⁺ → Algebra α ρᵃ
𝔄⁺ i = ∣ i ∣

𝐂 : Algebra ι ι
𝐂 = ⨅ 𝔄⁺

skEqual : (i : ℑ⁺) → ∀{p q} → Type ρᵃ
skEqual i {p}{q} = ⟦ p ⟧ ⟨$⟩ snd ∥ i ∥ ≈ ⟦ q ⟧ ⟨$⟩ snd ∥ i ∥
  where open Setoid 𝔻[ 𝔄⁺ i ] using ( _≈_ ) ; open Environment (𝔄⁺ i) using ( ⟦_⟧ )
```

The type skEqual provides a term identity p $\approx$ q for each index i = ($\mathbf{A}$, p , ρ) of the product. Later we prove that if the identity p $\approx$ q holds in all $\mathbf{A} \in$ S $\mathcal{K}$ (for all environments), then p $\approx$ q holds in the relatively free algebra $\mathbb{F}[\,\mathsf{X}\,]$; equivalently, the pair (p , q) belongs to the kernel of the natural homomorphism from $\mathbf{T}$ X onto $\mathbb{F}[\,\mathsf{X}\,]$. We will use that fact to prove that the kernel of the natural hom from $\mathbf{T}$ X to $\mathbf{C}$ is contained in the kernel of the natural hom from $\mathbf{T}$ X onto $\mathbb{F}[\,\mathsf{X}\,]$, whence we construct a monomorphism from $\mathbb{F}[\,\mathsf{X}\,]$ into $\mathbf{C}$, and thus $\mathbb{F}[\,\mathsf{X}\,]$ is a subalgebra of $\mathbf{C}$, so belongs to S (P $\mathcal{K}$).

```
homC : hom (𝐓 X) 𝐂
homC = ⨅-hom-co 𝔄⁺ (λ i → lift-hom (snd ∥ i ∥))

kerF⊆kerC : ker ∣ homF[ X ] ∣ ⊆ ker ∣ homC ∣
kerF⊆kerC {p , q} pKq (𝐀 , sA , ρ) = Goal
  where
  open Setoid 𝔻[ 𝐀 ]   using ( _≈_ ; sym ; trans )
  open Environment 𝐀 using ( ⟦_⟧ )
  fl : ∀ t → ⟦ t ⟧ ⟨$⟩ ρ ≈ free-lift ρ t
  fl t = free-lift-interp {𝐀 = 𝐀} ρ t

  ζ : ∀{p q} → (Th 𝒦) ⊢ X ▷ p ≈ q → 𝒦 ∣⊨ p ≈ q
  ζ x 𝐀 kA = sound (λ y ρ → y 𝐀 kA ρ) x where open Soundness (Th 𝒦) 𝐀

  subgoal : ⟦ p ⟧ ⟨$⟩ ρ ≈ ⟦ q ⟧ ⟨$⟩ ρ
  subgoal = S-id1{ℓ = ℓ}{p = p}{q} (ζ pKq) 𝐀 sA ρ
  Goal : (free-lift{𝐀 = 𝐀} ρ p) ≈ (free-lift{𝐀 = 𝐀} ρ q)
  Goal = trans (sym (fl p)) (trans subgoal (fl q))

homFC : hom 𝔽[ X ] 𝐂
```

```
homFC = | HomFactor 𝐂 homC homF[ X ] kerF⊆kerC (isSurjective ∥ epiF[ X ] ∥) |
```

If (p , q) belongs to the kernel of homC, then Th $\mathscr{K}$ includes the identity p ≈ q—that is, Th $\mathscr{K}$ ⊢ X ▷ p ≈ q. Equivalently, if the kernel of homC is contained in that of homF[X]. We formalize this fact as follows.

```
kerC⊆kerF : ∀{p q} → (p , q) ∈ ker | homC | → (p , q) ∈ ker | homF[ X ] |
kerC⊆kerF {p}{q} pKq = S𝒦|⊨→kerF (S𝒦|⊨ pqEqual)
  where
  S𝒦|⊨ : (∀ i → skEqual i {p}{q}) → S{β = α}{ρᵃ} ℓ 𝒦 |⊨ p ≈ q
  S𝒦|⊨ x 𝐀 sA ρ = x (𝐀 , sA , ρ)
  S𝒦|⊨→kerF : S{β = α}{ρᵃ} ℓ 𝒦 |⊨ p ≈ q → (p , q) ∈ ker | homF[ X ] |
  S𝒦|⊨→kerF x = hyp (S-id2{ℓ = ℓ}{p = p}{q} x)

  pqEqual : ∀ i → skEqual i {p}{q}
  pqEqual i = goal
    where
    open Environment (𝔄⁺ i) using ( ⟦_⟧ )
    open Setoid 𝔻[ 𝔄⁺ i ]     using ( _≈_ ; sym ; trans )
    goal : ⟦ p ⟧ ⟨$⟩ snd ∥ i ∥ ≈ ⟦ q ⟧ ⟨$⟩ snd ∥ i ∥
    goal = trans (free-lift-interp{𝐀 = ∣ i ∣}(snd ∥ i ∥) p)
          ( trans (pKq i)(sym (free-lift-interp{𝐀 = ∣ i ∣} (snd ∥ i ∥) q)))
```

We conclude that the homomorphism from 𝔽[X] to 𝐂 is injective, whence 𝔽[X] is (isomorphic to) a subalgebra of 𝐂.

```
monFC : mon 𝔽[ X ] 𝐂
monFC = | homFC | , isMon
  where
  isMon : IsMon 𝔽[ X ] 𝐂 | homFC |
  isHom isMon = ∥ homFC ∥
  isInjective isMon {p}{q} φpq = kerC⊆kerF φpq

F≤C : 𝔽[ X ] ≤ 𝐂
F≤C = mon→≤ monFC
```

Using the last result we prove that 𝔽[X] belongs to S (P $\mathscr{K}$). This requires one more technical lemma concerning the classes S and P; specifically, a product of subalgebras of algebras in a class is a subalgebra of a product of algebras in the class; in other terms, P (S $\mathscr{K}$) ⊆ S (P $\mathscr{K}$), for every class $\mathscr{K}$. We state and prove this in Agda as follows.

```
private a = α ⊔ ρᵃ ; oaℓ = ov (a ⊔ ℓ)

PS⊆SP : P (a ⊔ ℓ) oaℓ (S{β = α}{ρᵃ} ℓ 𝒦) ⊆ S oaℓ (P ℓ oaℓ 𝒦)
PS⊆SP {𝐁} (I , ( 𝒜 , sA , B≅⨅A )) = Goal
  where
  ℬ : I → Algebra α ρᵃ
  ℬ i = ∣ sA i ∣
  kB : (i : I) → ℬ i ∈ 𝒦
  kB i = fst ∥ sA i ∥
  ⨅A≤⨅B : ⨅ 𝒜 ≤ ⨅ ℬ
  ⨅A≤⨅B = ⨅-≤ λ i → snd ∥ sA i ∥
  Goal : 𝐁 ∈ S{β = oaℓ}{oaℓ}oaℓ (P {β = oaℓ}{oaℓ} ℓ oaℓ 𝒦)
  Goal = ⨅ ℬ , (I , (ℬ , (kB , ≅-refl))) , (≅-trans-≤ B≅⨅A ⨅A≤⨅B)
```

With this we can prove that 𝔽[X] belongs to S (P 𝒦).

```
SPF : 𝔽[ X ] ∈ S ι (P ℓ ι 𝒦)
SPF = | spC | , (fst ∥ spC ∥) , (≤-transitive F≤C (snd ∥ spC ∥))
  where
  psC : 𝐂 ∈ P (α ⊔ ρᵃ ⊔ ℓ) ι (S ℓ 𝒦)
  psC = 𝔍⁺ , (𝔄⁺ , ((λ i → fst ∥ i ∥) , ≅-refl))
  spC : 𝐂 ∈ S ι (P ℓ ι 𝒦)
  spC = PS⊆SP psC
```

Finally, we prove that every algebra in Mod (Th (V 𝒦)) is a homomorphic image of 𝔽[X].

```
  module _ {𝒦 : Pred(Algebra α ρᵃ) (α ⊔ ρᵃ ⊔ ov ℓ)} where
    private c = α ⊔ ρᵃ ⊔ ℓ ; ι = ov c
    open FreeAlgebra {χ = c}(Th 𝒦) using ( 𝔽[_] )

Var⇒EqCl : ∀ 𝐀 → 𝐀 ∈ Mod (Th (V ℓ ι 𝒦)) → 𝐀 ∈ V ℓ ι 𝒦
Var⇒EqCl 𝐀 ModThA = 𝔽[ 𝕌[ 𝐀 ] ] , (spFA , AimgF)
  where
  spFA : 𝔽[ 𝕌[ 𝐀 ] ] ∈ S{ι} ι (P ℓ ι 𝒦)
  spFA = SPF{ℓ = ℓ} 𝒦
  epiFIA : epi 𝔽[ 𝕌[ 𝐀 ] ] (Lift-Alg 𝐀 ι ι)
  epiFIA = F-ModTh-epi-lift{ℓ = ℓ} (λ {p q} → ModThA{p = p}{q})
  φ : Lift-Alg 𝐀 ι ι IsHomImageOf 𝔽[ 𝕌[ 𝐀 ] ]
  φ = epi→ontohom 𝔽[ 𝕌[ 𝐀 ] ] (Lift-Alg 𝐀 ι ι) epiFIA
  AimgF : 𝐀 IsHomImageOf 𝔽[ 𝕌[ 𝐀 ] ]
  AimgF = ∘-hom | φ | (from Lift-≅), ∘-IsSurjective _ _ ∥ φ ∥(fromIsSurjective (Lift-≅{𝐀 = 𝐀}))
```

ModTh-closure and Var⇒EqCl show that V 𝒦 = Mod (Th (V 𝒦)) holds for every class 𝒦 of S-algebras. Thus, every variety is an equational class. This completes the formal proof of Birkhoff's variety theorem.

7 Related work

There have been a number of efforts to formalize parts of universal algebra in type theory besides ours, most notably

1. In [5], Capretta formalized the basics of universal algebra in the Calculus of Inductive Constructions using the Coq proof assistant;
2. In [15], Spitters and van der Weegen formalized the basics of universal algebra and some classical algebraic structures, also in the Calculus of Inductive Constructions using the Coq proof assistant and promoting the use of type classes;
3. In [10] Gunther et al developed what seemed (prior to the agda-algebras library) to be the most extensive library of formalized universal algebra to date; like agda-algebras, that work is based on dependent type theory, is programmed in Agda, and goes beyond the Noether isomorphism theorems to include some basic equational logic; although the coverage is less extensive than that of agda-algebras, Gunther et al do treat *multi-sorted* algebras, whereas agda-algebras is currently limited to single sorted structures.
4. In [2], "Amato et al formalize multi-sorted algebras with finitary operators in UniMath. Limiting to finitary operators is due to the restrictions of the UniMath type theory, which does not have W-types nor user-defined inductive types. These restrictions also

prompt the authors to code terms as lists of stack machine instructions rather than trees" (quoting [1]).

5. In [11], "Lynge and Spitters formalize multi-sorted algebras in HoTT, also restricting to finitary operators. Using HoTT they can define quotients as types, obsoleting setoids. They prove three isomorphism theorems concerning sub- and quotient algebras. A universal algebra or varieties are not formalized" (quoting [1]).
6. In [1], Abel gives a new formal proof of the soundness theorem and Birkhoff's completeness theorem for multi-sorted algebraic structures.

Acknowledgments

This work would not have been possible without the wonderful Agda language and the Agda Standard Library, developed and maintained by The Agda Team [18]. Most of the content of this paper was generated from the literate Agda file HSP.lagda and the LaTeX 2$_\varepsilon$ file agda-hsp.tex (processed with the commands agda –latex and pdflatex), which are available in the agda-algebras GitHub repository [7]. The first author was supported by the CoCoSym Project under the ERC Consolidator Grant (ERC CoG), No. 771005.

References

1 Andreas Abel. Birkhoff's Completeness Theorem for multi-sorted algebras formalized in Agda. *CoRR*, abs/2111.07936, 2021. arXiv:2111.07936.

2 Gianluca Amato, Marco Maggesi, and Cosimo Perini Brogi. Universal algebra in UniMath. *CoRR*, abs/2102.05952, 2021. arXiv:2102.05952.

3 Clifford Bergman. *Universal Algebra: fundamentals and selected topics*, volume 301 of *Pure and Applied Mathematics (Boca Raton)*. CRC Press, Boca Raton, FL, 2012.

4 G Birkhoff. On the structure of abstract algebras. *Proceedings of the Cambridge Philosophical Society*, 31(4):433–454, Oct 1935.

5 Venanzio Capretta. Universal algebra in type theory. In *Theorem proving in higher order logics (Nice, 1999)*, volume 1690 of *Lecture Notes in Comput. Sci.*, pages 131–148. Springer, Berlin, 1999. doi:10.1007/3-540-48256-3_10.

6 William DeMeo. The Agda Universal Algebra Library. GitHub.com, 2020. Ver. 1.0.0; source code: https://gitlab.com/ualib/ualib.gitlab.io. URL: https://ualib.gitlab.io.

7 William DeMeo and Jacques Carette. The Agda Universal Algebra Library. GitHub.com, 2021. Ver. 2.0.0; source code: https://github.com/ualib/agda-algebras. doi:10.5281/zenodo.5730534.

8 Martín Hötzel Escardó. Introduction to Univalent Foundations of mathematics with Agda. https://www.cs.bham.ac.uk/~mhe/HoTT-UF-in-Agda-Lecture-Notes/, May 2019. Accessed on 30 Nov 2021.

9 Martín Hötzel Escardó. Introduction to Univalent Foundations of mathematics with Agda. *CoRR*, abs/1911.00580, 2019. arXiv:1911.00580.

10 Emmanuel Gunther, Alejandro Gadea, and Miguel Pagano. Formalization of universal algebra in Agda. *Electronic Notes in Theoretical Computer Science*, 338:147 – 166, 2018. The 12th Workshop on Logical and Semantic Frameworks, with Applications (LSFA 2017). doi:https://doi.org/10.1016/j.entcs.2018.10.010.

11 Andreas Lynge and Bas Spitters. Universal algebra in hott. In *Proceedings of the 25th International Conference on Types for Proofs and Programs (TYPES 2019)*, 2019. URL: http://www.ii.uib.no/~bezem/abstracts/TYPES_2019_paper_7.

12 John C. Mitchell. *Foundations for Programming Languages*. MIT Press, Cambridge, MA, USA, 1996. URL: https://mitpress.mit.edu/books/foundations-programming-languages.

13 nLab authors. Constructive Mathematics. http://ncatlab.org/nlab/show/constructive%20mathematics, March 2021. Revision 65.

14 nLab authors. Martin-Löf dependent type theory. http://ncatlab.org/nlab/show/Martin-L%C3%B6f%20dependent%20type%20theory, November 2021. Revision 28.

15 Bas Spitters and Eelis Van der Weegen. Type classes for mathematics in type theory. *CoRR*, abs/1102.1323, 2011. arXiv:1102.1323.

16 The Agda Team. *Agda Language Reference section on Axiom K*, 2021. URL: https://agda.readthedocs.io/en/v2.6.1/language/without-k.html.

17 The Agda Team. *Agda Language Reference section on Safe Agda*, 2021. URL: https://agda.readthedocs.io/en/v2.6.1/language/safe-agda.html#safe-agda.

18 The Agda Team. *The Agda Standard Library*, 2021. URL: https://github.com/agda/agda-stdlib.

19 The Agda Team. *Agda Tools Documentation section on Pattern matching and equality*, 2021. URL: https://agda.readthedocs.io/en/v2.6.1/tools/command-line-options.html#pattern-matching-and-equality.